

CAS for Physics Examples

Electrostatics, Magnetism, Circuits and
Mechanics of Charged Particles

Part 1

Vector- and Scalar Fields

Electric Field

Leon Magiera and Josef Böhm

Contents

| | | |
|----|--------------------------|----|
| | Introduction | 3 |
| I | Vector and Scalar Fields | 4 |
| II | Electric Field | 15 |

Introduction

This text is devoted to solving practical problems in the field of electricity and magnetism using computer algebra systems (CAS). The problems described usually appear in the programs of standard General Physics courses at university level (engineering and science). Some of them would also be suitable for high schools.

Maxima is a powerful tool for the manipulation of symbolic and numerical expressions, including differentiation, integration, vectors, matrices, . . . and so on. It also provides commands for plotting functions, curves and data in two and three dimensions. We chose *Maxima* because it is a free software package. It can be downloaded from its website, <http://maxima.sourceforge.net>, where documentation in several languages can also be found.

The use of *Maxima* significantly reduces the computational work and allows a student (or teacher) to concentrate on physical ideas (which is most important), rather than on the very time consuming technical side - performing the derivations by hand.

For a first try of *Maxima*, you may wish to try the examples in [First Steps with Maxima](#). Please don't understand this paper as a complete introduction in working with *Maxima*. *Maxima*-experts will certainly find ways to solve some problems in another way. *Maxima* is much more powerful than it is shown here. You can produce complex programs, your own libraries and much more.

Leon Magiera is physicist at the Wroclaw University of Science and Technology, Josef Böhm is a retired school teacher and founder of the *International DERIVE and CAS-TI User Group*. His work was putting all parts together in a common form and adding *DERIVE* and *TI-NspireCAS* treatments of some examples. *DERIVE* is off the market since several years but still widely used all over the world. *DERIVE* and *TI-NspireCAS* are running up to the latest operations systems. The *DERIVE* and *TI-NspireCAS* parts can be distinguished by another font set.

the Authors

All files are available on request.

We welcome any comments and suggestions regarding this book. You may contact us via e-mail at:

leon.magiera@wp.pl
nojo.boehm@pgv.at

In preparation:

2nd part: Magnetic Field

3rd part: Circuits

4th part: Mechanics of Charged Particles

For DERIVE User Group information goto www.austromath.at/dug/

I. Vector and Scalar Fields

Vector and scalar fields are very important concepts in physics. They are just respectively a vector and a scalar functions of space variable. Some differential operators are defined acting on those functions, which are in common use in physics, like: *gradient*, *divergence*, *rotation* and *Laplacian*.

We start with definitions using Cartesian coordinates x, y, z .

The *Gradient* of a scalar function Φ is the vector

$$\mathit{grad}\Phi = \left[\frac{\partial\Phi}{\partial x}, \frac{\partial\Phi}{\partial y}, \frac{\partial\Phi}{\partial z} \right]. \quad \text{I.1}$$

The *Divergence* of the vector field $\vec{A} = [A_x, A_y, A_z]$ is a scalar of the form

$$\mathit{div}\vec{A} = \frac{\partial A_x}{\partial x} + \frac{\partial A_y}{\partial y} + \frac{\partial A_z}{\partial z}. \quad \text{I.2}$$

The *Curl* of the vector field $\vec{A} = [A_x, A_y, A_z]$ is the following vector field

$$\mathit{curl}\vec{A} = \begin{vmatrix} \vec{i} & \vec{j} & \vec{k} \\ \frac{\partial}{\partial x} & \frac{\partial}{\partial y} & \frac{\partial}{\partial z} \\ A_x & A_y & A_z \end{vmatrix}. \quad \text{I.3}$$

The *Laplacian* of a scalar function Φ is a scalar of the form

$$\Delta\Phi = \frac{\partial^2\Phi}{\partial x^2} + \frac{\partial^2\Phi}{\partial y^2} + \frac{\partial^2\Phi}{\partial z^2}. \quad \text{I.4}$$

Now we try to solve some problems:

I.1 Calculate the following expressions:

- the *gradient* of the function $\Phi(r) = \frac{1}{r}$,
- $\mathit{div}(\vec{r})$,
- $\mathit{div}\left(\frac{\vec{r}}{r}\right)$,
- $\mathit{div}\left(\mathit{grad}\left(\frac{1}{r}\right)\right)$,
- $\mathit{curl}(\vec{r})$,

where r is the modulus of the position vector $\vec{r}(x, y, z)$,

Solutions:

We need loading the library "vector":

```

Problem I.1
a)
(%i1) Phi(r):=1/r$
(%i2) r_: [x,y,z]$
(%i3) r:=sqrt(r_.r_);
(%o3)  $\sqrt{z^2+y^2+x^2}$ 
(%i4) load("vector")$
(%i5) grad(Phi(r));
(%o5)  $\left[ -\frac{x}{(z^2+y^2+x^2)^{3/2}}, -\frac{y}{(z^2+y^2+x^2)^{3/2}}, -\frac{z}{(z^2+y^2+x^2)^{3/2}} \right]$ 

```

Note that result %o5 can be rewritten in a simple concise form:

$$\text{grad}(\Phi(r)) = -\frac{\vec{r}}{r^3}.$$

```

b)
(%i6) div(r_);
(%o6) 3
c)
(%i7) div(r_/r);
(%o7)  $\frac{3}{\sqrt{z^2+y^2+x^2}} - \frac{z^2}{(z^2+y^2+x^2)^{3/2}} - \frac{y^2}{(z^2+y^2+x^2)^{3/2}} - \frac{x^2}{(z^2+y^2+x^2)^{3/2}}$ 
(%i8) radcan(%);
(%o8)  $\frac{2}{\sqrt{z^2+y^2+x^2}}$ 

```

Some problems are too difficult for a computer program. The next example is too hard for many computer algebra systems. Maxima – and DERIVE as well – returns:

```

d)
(%i18) div(grad(1/r));
(%o18)  $\frac{3}{(z^2+y^2+x^2)^{3/2}} + \frac{3z^2}{(z^2+y^2+x^2)^{5/2}} + \frac{3y^2}{(z^2+y^2+x^2)^{5/2}} + \frac{3x^2}{(z^2+y^2+x^2)^{5/2}}$ 
(%i19) radcan(%);
(%o19) 0

```

This is the DERIVE result:

$$\#1: \text{DIV}\left(\text{GRAD}\left(\frac{1}{|[x, y, z]|}\right)\right) = 0$$

Remark: The above result is not correct. The correct result is proportional to the *Dirac* $\delta(r)$. Unfortunately, many computer algebra systems we tried did not return the correct result.

My last attempt was "asking" *MATHEMATICA* (www.wolframalpha.com) and see the answers:

$$\begin{aligned} \operatorname{div} \operatorname{grad} \left(\frac{1}{\sqrt{|x|^2 + |y|^2 + |z|^2}} \right) &= \frac{1}{(|x|^2 + |y|^2 + |z|^2)^{5/2}} \left(-|x| \frac{\partial^2 |x|}{\partial x^2} (|y|^2 + |z|^2) + \right. \\ &|x|^3 \left(-\frac{\partial^2 |x|}{\partial x^2} \right) + |x|^2 \left(2 \left(\frac{\partial |x|}{\partial x} \right)^2 - |y| \frac{\partial^2 |y|}{\partial y^2} - \left(\frac{\partial |y|}{\partial y} \right)^2 - |z| \frac{\partial^2 |z|}{\partial z^2} - \left(\frac{\partial |z|}{\partial z} \right)^2 \right) - \\ &|z|^2 \left(\left(\frac{\partial |x|}{\partial x} \right)^2 + \left(\frac{\partial |y|}{\partial y} \right)^2 + |z| \frac{\partial^2 |z|}{\partial z^2} - 2 \left(\frac{\partial |z|}{\partial z} \right)^2 \right) - \\ &|y|^2 \left(\left(\frac{\partial |x|}{\partial x} \right)^2 - 2 \left(\frac{\partial |y|}{\partial y} \right)^2 + |z| \frac{\partial^2 |z|}{\partial z^2} + \left(\frac{\partial |z|}{\partial z} \right)^2 \right) - |y| |z|^2 \frac{\partial^2 |y|}{\partial y^2} - |y|^3 \frac{\partial^2 |y|}{\partial y^2} \Big) \end{aligned}$$

(x: first Cartesian coordinate | y: second Cartesian coordinate | z: third Cartesian coordinate)

Input in another form delivers the δ -function, great.

$$\operatorname{div} \operatorname{grad} \left(\frac{1}{\sqrt{x^2 + y^2 + z^2}} \right) = -4 \pi \delta \left(\sqrt{x^2 + y^2 + z^2} \right)$$

(x: first Cartesian coordinate | y: second Cartesian coordinate | z: third Cartesian coordinate)

```
e)
(%i11) curl(r_);
(%o11) [0, 0, 0]
```

Exercise: Show that the divergence of the vector field of the form $\frac{c\vec{r}}{r}$, with c an arbitrary constant equal zero.

I.2 Prove the identity $\operatorname{grad} \left(\frac{\vec{a} \cdot \vec{r}}{r^3} \right) = -\operatorname{curl} \left(\frac{\vec{a} \times \vec{r}}{r^3} \right)$, where \vec{a} is an arbitrary, constant vector.

$\vec{r} = \vec{r}(x, y, z)$ is a position vector and r denotes its modulus.

Solution:

We add the definition of \vec{a} (which is $a_$) and use the “~”-character as operation sign for the vector product. We show that the difference of both sides simplifies to the zero vector.

```
(%i1) r_: [x, y, z] $
      a_: [ax, ay, az] $
      r: sqrt(r_.r_) $
.
(%i4) load("vect") $
.
(%i5) ev(express(grad((a_.r_)/r^3)), diff) - ev(express(-curl((a_~r_)/r^3)), diff);
(%o5) [
  3 ax      3 x (az z + ay y + ax x)  3 z (az x - ax z)  3 y (ax y - ay x)
  (z^2 + y^2 + x^2)^{3/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}
,
  3 ay      3 y (az z + ay y + ax x)  3 z (ay z - az y)  3 x (ax y - ay x)      3 az
  (z^2 + y^2 + x^2)^{3/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{3/2}
,
  3 z (az z + ay y + ax x)  3 y (ay z - az y)  3 x (az x - ax z)
  (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}      (z^2 + y^2 + x^2)^{5/2}
]
.
(%i6) radcan(%);
(%o6) [0, 0, 0]
```

We can proof the identity in a second way:

```
(%i8) is(equal(ev(express(grad((a_.r_)/r^3)), diff),
              ev(express(-curl((a_~r_)/r^3)), diff)));
(%o8) true
```

It is remarkable that the DERIVE proof is much shorter and much more elegant (two ways):

#1: $[r_ := [x, y, z], a_ := [ax, ay, az]]$

#2: $\text{GRAD}\left(\frac{a_ \cdot r_}{|r_|^3}\right) = - \text{CURL}\left(\frac{\text{CROSS}(a_ , r_)}{|r_|^3}\right) = \text{true}$

#3: $\text{GRAD}\left(\frac{a_ \cdot r_}{|r_|^3}\right) + \text{CURL}\left(\frac{\text{CROSS}(a_ , r_)}{|r_|^3}\right) = [0, 0, 0]$

Michel Beaudin from ETS Montreal produced a fine vector analysis library for TI-NspireCAS (https://cours.etsmtl.ca/seg/mbeaudin/documents/Kit_ETS_MB.tns) which allows performing the proof in one single line:

| | |
|--|------------------|
| $r_ := [x \ y \ z]; a_ := [ax \ ay \ az]$ | $[ax \ ay \ az]$ |
| $\text{grad}\left(\frac{\text{dotP}(a_ , r_)}{(\text{norm}(r_))^3}, [x \ y \ z]\right) = -\text{curl}\left(\frac{\text{crossP}(a_ , r_)}{(\text{norm}(r_))^3}, [x \ y \ z]\right)$ | [true true true] |

é■

I.3 Prove the identity $\text{curl}(\text{curl}(\vec{a})) = \text{grad}(\text{div}(\vec{a})) - \Delta\vec{a}$, where

$\vec{a} = \vec{a}(a_x(x, y, z), a_y(x, y, z), a_z(x, y, z))$ is a differentiable vector. The symbol Δ denotes the

Laplace operator $\Delta = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2}$.

Solution:

The identity can be rewritten in the equivalent form

$$\text{curl}(\text{curl}(\vec{a})) - \text{grad}(\text{div}(\vec{a})) + \Delta\vec{a} = [0, 0, 0].$$

```
(%i1) load("vector")$
(%i2) a_ := [ax(x, y, z), ay(x, y, z), az(x, y, z)]$
(%i3) curl(curl(a_)) - grad(div(a_)) + laplacian(a_);
(%o3) [0, 0, 0]
```

Exercise: \vec{a} and $\vec{b} = \vec{b}(b_x(x, y, z), b_y(x, y, z), b_z(x, y, z))$ are arbitrary differentiable vectors and $f = f(x, y, z)$ and $g = g(x, y, z)$ are arbitrary differentiable functions.

Prove the following identities:

- a) $\text{curl}(\text{grad}(f)) = [0, 0, 0]$
- b) $\text{grad}(f \cdot g) - f \cdot \text{grad}(g) - g \cdot \text{grad}(f) = [0, 0, 0]$,
- c) $\text{div}(\text{curl}(\vec{a})) = 0$,
- d) $\text{div}(\vec{a} \times \vec{b}) = \vec{b} \cdot \text{curl}(\vec{a}) - \vec{a} \cdot \text{curl}(\vec{b})$,
- e) $\text{div}(f \cdot \vec{a}) = f \cdot \text{div}(\vec{a}) + \vec{a} \cdot \text{grad}(f)$,
- f) $\text{curl}(f \cdot \vec{a}) = f \cdot \text{curl}(\vec{a}) + \text{grad}(f) \times \vec{a}$.

We will demonstrate two of the exercises from above, a) and f):

a)

```
(%i1) load("vect")$
(%i2) curl(grad(f(x,y,z)));
(%o2) 0
(%i3) load("vector")$
(%i4) curl(grad(f(x,y,z)));
(%o4) [0, 0, 0]
```

That's not fine. Both packages deliver different results. We know that the curl should be a vector. To be on the safe side, let's ask another CAS:

$$\text{curl}(\text{grad}(f(x,y,z)), [x \ y \ z]), [x \ y \ z])$$

$$\left[\frac{d}{dy} \left(\frac{d}{dz} (f(x,y,z)) \right) - \frac{d}{dz} \left(\frac{d}{dy} (f(x,y,z)) \right), \frac{d}{dz} \left(\frac{d}{dx} (f(x,y,z)) \right) - \frac{d}{dx} \left(\frac{d}{dz} (f(x,y,z)) \right), \frac{d}{dx} \left(\frac{d}{dy} (f(x,y,z)) \right) - \frac{d}{dy} \left(\frac{d}{dx} (f(x,y,z)) \right) \right]$$

#1: $f(x, y, z) :=$

#2: $\text{CURL}(\text{GRAD}(f(x, y, z))) = [0, 0, 0]$

Although TI-NspireCAS does not simplify the partial derivatives we see that we obtain the zero vector. DERIVE gives the correct answer.

f)

```
(%i1) load("vect")$
(%i2) a_:[ax(x,y,z),ay(x,y,z),az(x,yz)]$
(%i3) f_:f(x,y,z)$
(%i4) ev(express(curl(f_*a_)-f_*curl(a_)-(grad(f_)*a_)),diff);
(%o4) [0, -f(x,y,z) * (d/dx az(x,yz)) - f(x,y,z) * (d/dz ax(x,y,z) - d/dx az(x,yz)) +
f(x,y,z) * (d/dz ax(x,y,z)), -f(x,y,z) * (d/dx ay(x,y,z) - d/dy ax(x,y,z)) + f(x,y,z) *
(d/dx ay(x,y,z)) - f(x,y,z) * (d/dy ax(x,y,z))]
(%i5) ratsimp(%);
(%o5) [0, 0, 0]
```

In more advanced textbooks *cross product* and *curl* in the Cartesian coordinate system may be written using *Levi-Civita* tensor ε_{ijk} . This tensor is defined as $\varepsilon_{ijk} = \vec{e}_i (\vec{e}_j \times \vec{e}_k)$ ($i,j,k = 1,2,3$), where \vec{e}_i , ($i=1,2,3$) are the unit vectors

$$\vec{e}_1 = [1, 0, 0], \vec{e}_2 = [0, 1, 0], \vec{e}_3 = [0, 0, 1].$$

Using this notation, the components of the *cross product* of the vectors

$$\vec{A} = [A_x, A_y, A_z] \equiv [A_1, A_2, A_3], \vec{B} = [B_x, B_y, B_z] \equiv [B_1, B_2, B_3]$$

can be written as follows

$$(\vec{A} \times \vec{B})_i \equiv \sum_j \sum_k \varepsilon_{ijk} A_j B_k \quad (i=1,2,3).$$

The components of the *curl* of $\vec{B} = [B_1, B_2, B_3]$ are given by

$$(\text{curl}(\vec{B}))_i \equiv \sum_j \sum_k \varepsilon_{ijk} B_{k,j}, \quad (i=1,2,3)$$

where $B_{k,j}$ is the partial derivative of the k -th component of vector \vec{B} with respect to the j -th variable. For example $B_{3,2} \equiv \frac{\partial B_3}{\partial y}$.

I.4. Using the definitions above calculate:

- a) the components of the *cross product* of the following vectors:

$$\vec{A} = [A_x, A_y, A_z], \quad \vec{B} = [B_x, B_y, B_z],$$

- b) the components of the *curl* of the vector

$$\vec{A} = [A_x(x, y, z), A_y(x, y, z), A_z(x, y, x)].$$

Solution:

We enter the unit vectors collected in an identity matrix and the *Levi-Civita* tensor:

```
(%i1) load("vect")$
(%i2) e:ident(3)$
      a_: [a1, a2, a3]$
      b_: [b1, b2, b3]$
(%i5) eps(i, j, k) := e[i] . (e[j] ~ e[k]);
(%o5) eps(i, j, k) := e_i . e_j ~ e_k
(%i6) express(eps(1, 2, 3) . a_[2] . b_[3]);
(%o6) a2 . b3
(%i7) express(eps(1, 2, 2) . a_[2] . b_[2]);
(%o7) [0, 0, 0] . a2 . b2
```

%o7 is not correct. Why that? Further investigation shows that the vector product $\vec{v} \times \vec{v}$ results in 0 instead of the zero vector.

```
(%i8) express([0, 1, 1] ~ [0, 1, 2]);
(%o8) [1, 0, 0]
(%i9) express([0, 1, 1] ~ [0, 1, 1]);
(%o9) 0
```

What to do? Easy enough: define your own cross product!

```
(%i16) crossp(va, vb) := [va[2]*vb[3]-va[3]*vb[2],
                        va[3]*vb[1]-va[1]*vb[3], va[1]*vb[2]-va[2]*vb[1]];
(%o16) crossp(va, vb) := [va_2 vb_3 - va_3 vb_2, va_3 vb_1 - va_1 vb_3, va_1 vb_2 - va_2 vb_1]
(%i11) eps_(i, j, k) := e[i] . crossp(e[j], e[k]);
(%o11) eps_(i, j, k) := e_i . crossp(e_j, e_k)
(%i12) makelist(sum(sum(eps_(i, j, k) * a_[j] * b_[k], j, 1, 3), k, 1, 3), i, 3);
(%o12) [a2 b3 - a3 b2, a3 b1 - a1 b3, a1 b2 - a2 b1]
```

Now it is correct. We proceed with problem b):

```

(%i13) aa_:=[ax(x,y,z),ay(x,y,z),az(x,y,z)]$
(%i14) r_:=[x,y,z]$
(%i15) makelist(sum(sum(eps_(i,j,k)*diff(aa_[k],r_[j])),j,1,3),k,1,3),i,3);
(%o15) [ $\frac{d}{dy}az(x,y,z) - \frac{d}{dz}ay(x,y,z)$ ,  $\frac{d}{dz}ax(x,y,z) - \frac{d}{dx}az(x,y,z)$ ,  $\frac{d}{dx}ay(x,y,z) - \frac{d}{dy}ax(x,y,z)$ ]

```

It's no problem at all solving I.4. with TI-NspireCAS:

The screenshot shows the TI-NspireCAS interface with the following steps:

- Define the identity tensor: $e_ := \text{identity}(3)$
- Define the vector field: $aa_ := [ax(x,y,z), ay(x,y,z), az(x,y,z)]$
- Define the position vector: $r_ := [x, y, z]$
- Calculate the epsilon tensor: $\epsilon(i,j,k) := \text{dotP}(e_[i], \text{crossP}(e_[j], e_[k]))$
- Define vectors a_ and b_: $a_ := [a1, a2, a3]$; $b_ := [b1, b2, b3]$
- Calculate the dot product of a_ and b_: $\{a2 \cdot b3 - a3 \cdot b2, a3 \cdot b1 - a1 \cdot b3, a1 \cdot b2 - a2 \cdot b1\}$
- Calculate the curl components using the epsilon tensor: $\text{seq}\left(\sum_{k=1}^3 \left(\sum_{j=1}^3 (\epsilon(i,j,k) \cdot a_{[1,j]} \cdot b_{[1,k]})\right), i, 1, 3\right)$
- Define the vector field aa_ and position vector r_ again: $aa_(x,y,z) := [ax(x,y,z), ay(x,y,z), az(x,y,z)]$; $r_ := [x, y, z]$
- Calculate the curl of aa_ using the epsilon tensor: $\text{seq}\left(\sum_{k=1}^3 \left(\sum_{j=1}^3 (\epsilon(i,j,k) \cdot \frac{d}{dr_{[1,j]}}(aa_(x,y,z)[1, k]))\right), i, 1, 3\right)$
- Calculate the curl of aa_ using the built-in function: $\text{kit_ets_mb}\backslash\text{curl}(aa_(x,y,z), [x, y, z])$

Exercise: Using the tensor ϵ_{ijk} calculate $\text{curl}(\text{curl } \vec{A})$ and compare the result with the result obtained with the help of the built-in function `curl`.

In many problems it is convenient to apply a curvilinear coordinate system, like a spherical or a cylindrical one, instead of the *Cartesian* system.

In such a system, the *gradient* of the function f is the vector

$$\text{grad}(f) = \sum_i \frac{1}{s_i} \frac{\partial f}{\partial \alpha_i} \vec{n}_i \quad (\text{II.1a})$$

where the s_i are *Lame*^[*]-coefficients and can be written as

$$s_i = \sqrt{\left(\frac{\partial x}{\partial \alpha_i}\right)^2 + \left(\frac{\partial y}{\partial \alpha_i}\right)^2 + \left(\frac{\partial z}{\partial \alpha_i}\right)^2} \dots \quad (\text{II.2a})$$

x, y, z are the Cartesian coordinates and

$$(\alpha_1, \alpha_2, \alpha_3)$$

denote curvilinear coordinates and \vec{n}_i denote the local unit vectors.

The *divergence* of the vector field $\vec{A}(A_1, A_2, A_3)$ is the following scalar quantity

$$\text{div}\vec{A} = \frac{1}{s_1 s_2 s_3} \left[\frac{\partial(s_2 s_3 A_1)}{\partial \alpha_1} + \frac{\partial(s_1 s_3 A_2)}{\partial \alpha_2} + \frac{\partial(s_1 s_2 A_3)}{\partial \alpha_3} \right] \quad (\text{II.3a})$$

The *curl* of a vector field $\vec{A}(A_1, A_2, A_3)$ is of the form

$$\text{curl}(\vec{A}) = \begin{vmatrix} \vec{n}_1 & \vec{n}_2 & \vec{n}_3 \\ s_2 s_3 & s_1 s_3 & s_1 s_2 \\ \frac{\partial}{\partial \alpha_1} & \frac{\partial}{\partial \alpha_2} & \frac{\partial}{\partial \alpha_3} \\ s_1 A_1 & s_2 A_2 & s_3 A_3 \end{vmatrix} \quad (\text{II.4a})$$

The *Laplacian* of the function f may be written as

$$\Delta f = \frac{1}{s_1 s_2 s_3} \left[\frac{\partial \left(\frac{s_2 s_3}{s_1} \frac{\partial f}{\partial \alpha_1} \right)}{\partial \alpha_1} + \frac{\partial \left(\frac{s_1 s_3}{s_2} \frac{\partial f}{\partial \alpha_2} \right)}{\partial \alpha_2} + \frac{\partial \left(\frac{s_1 s_2}{s_3} \frac{\partial f}{\partial \alpha_3} \right)}{\partial \alpha_3} \right] \quad (\text{II.5a})$$

To solve the next problems we will use the definitions from above, as well as the built-functions of the computer algebra systems.

^[1] *Gabriel Lamé*, French mathematician 1795-1870. Well known is the *Lamé-Curve* $\left(\frac{x}{a}\right)^n + \left(\frac{y}{b}\right)^n = 1$ which with $n > 1$ is also called "super ellipse".

I.5. Determine the *gradient* of the generic function $f(r,\theta,\varphi)$ defined in spherical coordinates.

Solution:

We enter the Cartesian and the spherical coordinates and the relations between them, followed by the definition of the *Lamé* coefficients. Applying `trigreduce` on I.1a gives the *gradient* – but

not in its most comfortable form. The second component can be rewritten as $\frac{d}{d\varphi} f(r,\varphi,\theta)$
 $\frac{d}{d\varphi} f(r,\varphi,\theta)$
 $r \cdot \sin(\theta)$.

```
(%i1) x:r*sin(theta)*cos(phi)$
      y:r*sin(theta)*sin(phi)$
      z:r*cos(theta)$

(%i4) r_: [x,y,z]$
      alpha_: [r,phi,theta]$

(%i6) s(j):=sqrt(sum(diff(r_[i],alpha_[j])^2,i,1,3));

(%o6) s(j):=sqrt(sum_{i=1}^3 diff(r_{-i},alpha_{-j})^2)

(%i7) trigreduce(makelist(1/s(j)*diff(f(r,phi,theta),alpha_[j]),j,3));

(%o7) [frac{d}{dr} f(r, phi, theta), sqrt(2) * (frac{d}{d phi} f(r, phi, theta)) / sqrt(r^2 - r^2 cos(2 theta)), frac{d}{d theta} f(r, phi, theta) / |r|]

(%i8) trigreduce(sqrt(1-cos(2*theta)));

(%o8) sqrt(1-cos(2 theta))

(%i9) trigexpand(%o8);

(%o9) sqrt(sin(theta)^2 - cos(theta)^2 + 1)

(%i10) trigsimp(%o9);

(%o10) sqrt(2) |sin(theta)|
```

DERIVE enables the same procedure. It is not necessary to manipulate the result in order to obtain the final result. DERIVE offers a second – much shorter – method using an extended GRAD-function:

#1: $f(r, \varphi, \theta) :=$

$$\#2: \text{GRAD}(f(r, \varphi, \theta), \text{spherical}) = \begin{bmatrix} \frac{d}{dr} f(r, \varphi, \theta), & \frac{d}{d\theta} f(r, \varphi, \theta) / (r \cdot \text{SIN}(\varphi)), & \frac{d}{d\varphi} f(r, \varphi, \theta) / r \end{bmatrix}$$

`spherical` is the coordinate geometry matrix with the independent variables in the first row and the corresponding *Lamé* coefficients in the second one.

Comment: In case of the right-hand coordinate system, in which the positive z-axis is associated with the north pole, the colatitude φ is measured in radians south from the north pole and colongitude θ is measured in radians east from the positive x-axis. In the DERIVE function the angles φ and θ have the reverse meaning. It means that φ is replaced by θ and θ by φ , which can be seen when we compare results %o7 (wxMaxima) and #2 (DERIVE).

Let's have a look on the TI-NspireCAS screen:

The TI-NspireCAS screen displays the following content:

- $r_ := [x \ y \ z] : a_ := [r \ \varphi \ \theta]$
- $x := r \cdot \sin(\theta) \cdot \cos(\varphi) : y := r \cdot \sin(\theta) \cdot \sin(\varphi) : z := r \cdot \cos(\theta)$
- $s(j) := \sqrt{\sum_{i=1}^3 \left(\frac{d}{da_{[1,j]}}(r_{[1,i]}) \right)^2}$
- $r_ := [x \ y \ z] : a_ := [r \ \varphi \ \theta]$
- $x := r \cdot \sin(\theta) \cdot \cos(\varphi) : y := r \cdot \sin(\theta) \cdot \sin(\varphi) : z := r \cdot \cos(\theta)$
- $s(j) := \sqrt{\sum_{i=1}^3 \left(\frac{d}{da_{[1,j]}}(r_{[1,i]}) \right)^2}$
- $\text{list} \blacktriangleright \text{mat} \left(\text{seq} \left(\frac{1}{s(j)} \cdot \frac{d}{da_{[1,j]}}(f(r, \varphi, \theta)), j, 1, 3 \right), 3 \right) | r > 0 \text{ and } 0 \leq \theta \leq \pi$
- $\left[\begin{array}{ccc} \frac{d}{dr}(f(r, \varphi, \theta)) & \frac{d}{d\varphi}(f(r, \varphi, \theta)) & \frac{d}{d\theta}(f(r, \varphi, \theta)) \\ \frac{d}{dr}(f(r, \varphi, \theta)) & \frac{d}{d\varphi}(f(r, \varphi, \theta)) & \frac{d}{d\theta}(f(r, \varphi, \theta)) \\ \sin(\theta) \cdot r & r & r \end{array} \right]$
- © the Lamé coefficients for spherical coordinates are:
- $\text{seq}(s(j), j, 1, 3) | r > 0 \text{ and } 0 \leq \theta \leq \pi$
- $\{ 1, \sin(\theta) \cdot r, r \}$

Exercise: First generate the geometry matrices for Cartesian, spherical and cylindrical coordinates and then write a generalized function for calculating the gradient for all systems.



I.6. Calculate:

- the *divergence* and
- the *curl* of an arbitrary vector given in the spherical coordinate system:

$$\vec{A} = [A_r(r, \varphi, \theta), A_\varphi(r, \varphi, \theta), A_\theta(r, \varphi, \theta)]$$

Solution:

We start with problem a) and proceed calculating the *curl*:

```
(%i1) x:r*sin(theta)*cos(phi)$ y:r*sin(theta)*sin(phi)$ z:r*cos(theta)$
      r_: [x,y,z]$ a_: [r,phi,theta]$
(%i6) aa_: [ar(r,phi,theta), aph(r,phi,theta), ath(r,phi,theta)];
(%o6) [ar(r, phi, theta), aph(r, phi, theta), ath(r, phi, theta)]
(%i7) s(j) := sqrt(sum(diff(r_[i], a_[j])^2, i, 1, 3));
(%o7) s(j) := \sqrt{\sum_{i=1}^3 \text{diff}(r_{-i}, a_{-j})^2}
a) the divergence:
(%i8) assume(r>0)$
(%i9) trigsimp(1/(s(1)*s(2)*s(3))*(diff(s(2)*s(3)*aa_[1],
a_[1])+diff(s(1)*s(3)*aa_[2], a_[2])+diff(s(1)*s(2)*aa_[3], a_[3])));
(%o9) \left(\frac{d}{d\varphi} \text{aph}(r, \varphi, \theta)\right) |\sin(\theta)| + \left(\frac{d}{d\theta} \text{ath}(r, \varphi, \theta) + r \left(\frac{d}{dr} \text{ar}(r, \varphi, \theta)\right) + 2 \text{ar}(r, \varphi, \theta)\right)
\sin(\theta)^2 + \text{ath}(r, \varphi, \theta) \cos(\theta) \sin(\theta) / (r \sin(\theta)^2)
(%i10) expand(%);
(%o10) \frac{\frac{d}{d\varphi} \text{aph}(r, \varphi, \theta)}{r |\sin(\theta)|} + \frac{\text{ath}(r, \varphi, \theta) \cos(\theta)}{r \sin(\theta)} + \frac{\frac{d}{d\theta} \text{ath}(r, \varphi, \theta)}{r} + \frac{\frac{d}{dr} \text{ar}(r, \varphi, \theta)}{r} + \frac{2 \text{ar}(r, \varphi, \theta)}{r}
```

```
b) the components of the curl:
(%i11) expand(trigsimp(1/(s(2)*s(3))*(diff(s(3)*aa_[3], a_[2]) - diff(s(2)*aa_[2], a_[3]))));
(%o11) \frac{\frac{d}{d\varphi} \text{ath}(r, \varphi, \theta)}{r |\sin(\theta)|} - \frac{\text{aph}(r, \varphi, \theta) \cos(\theta)}{r \sin(\theta)} - \frac{\frac{d}{d\theta} \text{aph}(r, \varphi, \theta)}{r}
(%i12) expand(trigsimp(1/(s(1)*s(3))*(diff(s(1)*aa_[1], a_[3]) - diff(s(3)*aa_[3], a_[1]))));
(%o12) -\frac{\frac{d}{dr} \text{ath}(r, \varphi, \theta)}{r} + \frac{\frac{d}{d\theta} \text{ar}(r, \varphi, \theta)}{r} - \frac{\text{ath}(r, \varphi, \theta)}{r}
(%i13) expand(trigsimp(1/(s(1)*s(2))*(diff(s(2)*aa_[2], a_[1]) - diff(s(1)*aa_[1], a_[2]))));
(%o13) -\frac{\frac{d}{d\varphi} \text{ar}(r, \varphi, \theta)}{r |\sin(\theta)|} + \frac{\frac{d}{dr} \text{aph}(r, \varphi, \theta)}{r} + \frac{\text{aph}(r, \varphi, \theta)}{r}
```

You can work with DERIVE or with TI-NspireCAS as demonstrated above (one example):

$$\#10: \frac{1}{s(2) \cdot s(3)} \cdot \left(\frac{\partial}{\partial \alpha_3} (s(3) \cdot (A_r(r, \varphi, \theta)))_{\alpha_2} - \frac{\partial}{\partial \alpha_2} (s(2) \cdot (A_r(r, \varphi, \theta)))_{\alpha_3} \right)$$

$$\#11: \frac{\frac{d}{d\varphi} A_\theta(r, \varphi, \theta)}{r \cdot |\sin(\theta)|} - \frac{A_\varphi(r, \varphi, \theta) \cdot \cot(\theta)}{r} - \frac{\frac{d}{d\theta} A_\varphi(r, \varphi, \theta)}{r}$$

Exercise: Calculate the *divergence* of an arbitrary vector in cylindrical coordinates.

$$\vec{A} = [A_r(r, \varphi, z), A_\varphi(r, \varphi, z), A_z(r, \varphi, z)].$$

I.7. Find the expression for the *Laplacian* of the function $\Phi(r, \varphi, \theta)$ in spherical coordinates.

Solution:

We enter definition (I.5a) in a more concise form:

$$\frac{1}{\prod_{j=1}^3 s(j)} \cdot \sum_{i=1}^3 \left(\frac{\partial}{\partial \alpha_i} \left(\frac{\prod_{k=1}^3 s(k)}{s(i)^2} \cdot \frac{\partial}{\partial \alpha_i} f(r, \varphi, \theta) \right) \right)$$

```
(%i14) x:r*sin(theta)*cos(phi)$
      y:r*sin(theta)*sin(phi)$
      z:r*cos(theta)$
      r_: [x,y,z]$
      a_: [r,phi,theta]$

(%i19) s(j) := sqrt(sum(diff(r_[i], a_[j])^2, i, 1, 3));

(%o19) s(j) := \sqrt{\sum_{i=1}^3 \text{diff}(r_{-i}, a_{-j})^2}

(%i20) trigsimp(1/product(s(j), j, 1, 3) * sum(diff(product(s(k), k, 1, 3) /
      s(i)^2 * diff(f(r, phi, theta), a_[i]), a_[i]), i, 1, 3));

(%o20) \left( \frac{d^2}{d\theta^2} f(r, \varphi, \theta) + r^2 \left( \frac{d^2}{dr^2} f(r, \varphi, \theta) \right) + 2r \left( \frac{d}{dr} f(r, \varphi, \theta) \right) \right) \sin(\theta)^2 +
      \left( \frac{d}{d\theta} f(r, \varphi, \theta) \right) \cos(\theta) \sin(\theta) + \frac{d^2}{d\varphi^2} f(r, \varphi, \theta) / (r^2 \sin(\theta)^2)
```

We can compare with the DERIVE result (consider that φ and θ are interchanged!):

LOAD(C:\Program Files\TI Education\Derive 6\Math\VectorMatrixFunctions.mth)

f(r, phi, theta) :=

LAPLACIAN(f(r, phi, theta), spherical)

$$\frac{\cot(\varphi) \cdot \frac{d}{d\varphi} f(r, \varphi, \theta)}{r^2} + \frac{\left(\frac{d}{d\theta} \right)^2 f(r, \varphi, \theta)}{r^2 \cdot \sin(\varphi)^2} + \frac{r^2 \cdot \left(\frac{d}{dr} \right)^2 f(r, \varphi, \theta) + 2 \cdot r \cdot \frac{d}{dr} f(r, \varphi, \theta) + \left(\frac{d}{d\varphi} \right)^2 f(r, \varphi, \theta)}{r^2}$$

Exercise: Find the *Laplacian* of function $f(r, \varphi, z)$ in cylindrical coordinates.


II. Electric Field

Introduction

The most elementary part of physics dealing with electromagnetic interactions is called electrostatics. Electrostatics describes the interaction between stationary electric charges. The force of interaction between two point charges obeys *Coulomb's law*. Force \vec{F} acting on a charge q whose position is \vec{r} , due to a charge Q at \vec{R} , is given by

$$\vec{F} = \frac{1}{4\pi\epsilon_0} \frac{qQ}{|\vec{r} - \vec{R}|^3} (\vec{r} - \vec{R}) ,$$

where ϵ_0 denotes the permittivity in a vacuum.



Electrostatic interactions may also be described in a different way, by means of the concept of an electric field. Within this formalism, the force \vec{F} acting on a charge q is the product of the charge q and the electric field \vec{E} caused by the charge Q

$$\vec{F} = q \cdot \vec{E},$$

hence

$$\vec{E} = \frac{\vec{F}}{q} = \frac{1}{4\pi\epsilon_0} \frac{Q}{|\vec{r} - \vec{R}|^3} (\vec{r} - \vec{R}) .$$

The modern formulation of electrostatics is based on *Gauss's law*^{*}, which may be written in the form

$$\oint_S \vec{E} \cdot d\vec{S} = \frac{Q}{\epsilon_0},$$

where Q is the total charge giving rise to the electrostatic field \vec{E} , confined by a closed surface S . In differential notation *Gauss's law* is of the form

$$\text{div}(\vec{E}) = \frac{\rho}{\epsilon_0},$$

where ρ is the charge density.

One very important property of an electrostatic field is the fact that it can be described by a scalar function called the *potential*. This is a consequence of conservative nature of electrostatic force. Field intensity \vec{E} and potential Φ are related to each other by the equation

$$\vec{E} = -\text{grad}(\Phi).$$

(*) https://en.wikipedia.org/wiki/Gauss's_law

Thus, *Gauss's* law, expressed in terms of the potential, reads

$$\nabla^2\Phi = -\frac{\rho}{\epsilon_0},$$

which is known as *Poisson's* law (∇^2 denotes the *Laplace* operator).

The electric field created by a system of electric charges is the vector sum of the fields created by the individual charges (the principle of superposition).

Using the equations given above, we can solve standard exercises as presented below. Some of them will deal with discrete charges, the other ones with continuously distributed charges.

PROBLEMS

II.1 Split point charge Q into two point charges q and $(Q - q)$, separated by a distance of d , so that the force of interaction between them is maximum.

Solution: According to *Coulomb's* law, the magnitude of the force of repulsion between the charges is given by

```
(%i1) F:1/(4*pi*epsilon[0])*q*(Q-q)/d^2;
(%o1)  q(Q-q)
      4 pi epsilon_0 d^2
```

and

```
(%i2) solve(diff(F,q)=0,q);
(%o2) [ q=Q/2]
```

The problem can also be solved by directly finding the maximum value of the function F (even its nominator only) i.e. by solving the equation

$$\frac{d}{dq}q(Q - q) = 0.$$

Let's find the solution of the above equation

```
(%i3) solve(diff(q*(Q-q),q)=0,q);
(%o3) [ q=Q/2]
```

and then check the kind of the turning point by evaluating the second derivative for the obtained solution

```
(%i5) sign(diff(q*(Q-q),q,2));
(%o5) neg
```

From %o3 and %o5 we see that at $q = Q/2$ the interactive force is maximum.

Note: In the above exercise evaluation of sign function is not necessary. See below

```
(%i6) diff(q*(Q-q),q,2);
(%o6) -2
```

We will skip the DERIVE and the TI-NspireCAS treatment as well because we recognize the quadratic function with zeros at $q_1 = 0$ and $q_2 = Q$. As the graph is a downwards open parabola its vertex is at $q_0 = Q/2$.

II.2 We consider a system of an infinite number of point charges Q , which are distributed on a half-line in such a way that the first charge is put at a distance of d from the beginning of the half-line, A , and each successive distance is chosen to be as α times greater than the previous one. Find the force acting on a point charge q placed at A .

Solution: The resultant force acting on q is the algebraic sum of *Coulomb* forces originating from the successive charges Q .

```
(%i1) F:q*Q/(4*pi*epsilon[0])*sum(1/(alpha^n*d)^2,n,0,inf);
```

$$\frac{\left(\sum_{n=0}^{\infty} \frac{1}{\alpha^{2n}} \right) q Q}{4 \pi \epsilon_0 d^2}$$

```
(%o1)
```

Let's try to evaluate the above expression.

```
(%i2) assume(alpha>0) $
(%i3) 'F=F,simpsum;
Is alpha-1 positive, negative, or zero?p;
```

$$\frac{\left(\sum_{n=0}^{\infty} \frac{1}{\alpha^{2n}} \right) q Q}{4 \pi \epsilon_0 d^2} = \frac{q Q}{4 \pi \epsilon_0 \left(1 - \frac{1}{\alpha^2} \right) d^2}$$

```
(%o3)
```

The above sum was calculated with $\alpha - 1 > 0$ i.e. the force takes the simple form for $\alpha > 1$.

For $\alpha < 1$ we get

```
(%i4) 'F=F,simpsum;
Is alpha-1 positive, negative, or zero?n;
sum: sum is divergent.
-- an error. To debug this try: debugmode(true)
```

Let us notice that in our series (geometrical series) the ratio $\frac{1}{\alpha^2}$ appears.

```
(%i5) a[n]:=1/alpha^(2*n) $
(%i6) expand(
        a[n+1]/a[n]
    );
```

$$\frac{1}{\alpha^2}$$

```
(%o6)
```

Hence the solution of our problem ($\alpha > 1$).

This is the DERIVE solution:

#1: [CaseMode := Sensitive, InputMode := Word]

$$\#2: F := \frac{q \cdot Q}{4 \cdot \pi \cdot \epsilon_0} \cdot \sum_{n=0}^{\infty} \frac{1}{(\alpha \cdot d)^2}$$

$$\#3: F := \frac{Q \cdot q \cdot \sum_{n=0}^{\infty} \alpha^{-2 \cdot n}}{4 \cdot \pi \cdot d^2 \cdot \epsilon_0}$$

We see that there is a geometric series with quotient $1/\alpha^2$.

$$\#4: \frac{\alpha^{-2 \cdot (n+1)}}{\alpha^{-2 \cdot n}} = \frac{1}{\alpha^2}$$

#5: $\alpha \in \text{Real}(1, \infty)$

F from above is recalculated and is given in a simpler form:

$$\#6: F := \frac{Q \cdot q \cdot \alpha^2}{4 \cdot \pi \cdot d^2 \cdot \epsilon_0 \cdot (\alpha^2 - 1)}$$

Set $\alpha \leq 1$:

#7: $\alpha \in \text{Real}(-\infty, 1]$

$$\#8: F := \frac{Q \cdot q \cdot \sum_{n=0}^{\infty} \alpha^{-2 \cdot n}}{4 \cdot \pi \cdot d^2 \cdot \epsilon_0}$$

With TI-NspireCAS we cannot distinguish between lower and uppercase characters, so we denote Q as qq:

$$\frac{q \cdot qq}{4 \cdot \pi \cdot \epsilon_0} \cdot \sum_{n=0}^{\infty} \left(\frac{1}{(\alpha^n \cdot d)^2} \right) \rightarrow \frac{\sum_{n=0}^{\infty} \left(\frac{\alpha^{-2 \cdot n}}{d^2} \right) \cdot q \cdot qq}{4 \cdot \epsilon_0 \cdot \pi}$$

$$\frac{q \cdot qq}{4 \cdot \pi \cdot \epsilon_0} \cdot \sum_{n=0}^{\infty} \left(\frac{1}{(\alpha^n \cdot d)^2} \right) \Big|_{\alpha > 1} \rightarrow \frac{q \cdot qq \cdot \alpha^2}{4 \cdot d^2 \cdot (\alpha^2 - 1) \cdot \epsilon_0 \cdot \pi}$$


II.3 Three stationary charges Q_1, Q_2 and Q_3 are placed at the following positions $\vec{R}_1(x_1, y_1, z_1)$, $\vec{R}_2(x_2, y_2, z_2)$ and $\vec{R}_3(x_3, y_3, z_3)$, respectively.

- Write down expressions describing the field potential at the point $\vec{r}(x, y, z)$.
- Compute the resultant force acting on an electron of charge q for the data:
 $x = 6\text{m}, y = -7\text{m}, z = 9\text{m}, Q_1 = 4\text{C}, Q_2 = -3\text{C}, Q_3 = 0.5\text{C}$,
 $\vec{R}_1(0, 2, 0), \vec{R}_2(-3, 2, 0)$ and $\vec{R}_3(0, 0, 0)$.

Solution:

The potential at the point \vec{r} , due to a single charge Q_j placed at the point given by the position vector \vec{R}_j can be expressed as

$$\Phi_j = \frac{1}{4\pi\epsilon_0} \frac{Q_j}{|\vec{r} - \vec{R}_j|}$$



According to the principle of superposition, the resultant potential is

$$\Phi = \sum_j \Phi_j,$$

a) To make the notation more concise the components of the position vectors \vec{R}_j of the charges Q_j are entered in the form of the matrix \mathbf{R}_- .

```
(%i1) R_:=matrix([0,2,0],[-3,2,0],[0,0,0])*m;
(%o1) [ 0  2 m  0
      -3 m  2 m  0
      0   0  0 ]
```

while the charges as components of the vector \mathbf{Q}_- .

```
(%i2) Q_:[4,-3,1/2]*C;
(%o2) [ 4 C, -3 C, C/2 ]
```

Then the resultant potential at $\vec{r}(x,y,z)$ can be written as follows:

```
(%i3) Phi(x,y,x):=1/(4*pi*epsilon[0])*
      sum(Q_[j]/sqrt((r_-R_[j]).(r_-R_[j])),j,1,length(Q_));
(%o3) Phi(x,y,x):=1/(4*pi*epsilon_0) * sum_{j=1}^{length(Q_-)} Q_{-j} / sqrt((x_-R_{-j}) . (x_-R_{-j}))
```

with

```
(%i4) r_:[x,y,x];
(%o4) [ x, y, x ]
```

b) The desired force has the form

$$F = \sqrt{\vec{F} \cdot \vec{F}}$$

where

$$\vec{F} = q \cdot \vec{E}, \quad \vec{E} = -\text{grad}(\Phi)$$

We load the file “vect” and enter the above relations.

```
(%i5) load(vect)$
(%i6) E_(x,y,z):=-ev(express(grad(Phi(x,y,z))),diff)$
(%i7) F_(x,y,z):=q*E_(x,y,z)$
```

The magnitude of the resultant force is

```
(%i8) F(x,y,z):=sqrt(F_(x,y,z).F_(x,y,z))$
```

We import the value of the electron charge and electric constant from the utility file “physical_constants”.

```
(%i9) load(physical_constants)$
      propvars(physical_constant);
(%o10) [%c,%mu_0,%e_0,%Z_0,%G,%h,%h_bar,%m_P,%T_P,%l_P,%t_P,%e,%Phi_0,%G_0,%K_J,%R_K,
%mu_B,%mu_N,%a,%R_inf,%a_0,%E_h,%ratio_h_me,%m_e,%N_A,%m_u,%F,%R,%k,%V_m,%n_0,%ratio_SO,
%sigma,%c_1,%c_1L,%c_2,%b,%h_prime]
(%i11) get(%e,description);
(%o11) elementary charge
(%i12) get(%e_0,description);
(%o12) electric constant $1/(\mu_0 c^2)$
```

In the next step we evaluate the magnitude of the resultant force at the point $\vec{r}[6\text{m}, -7\text{m}, 9\text{m}]$.

```
(%i13) float(subst([q=constvalue(%e),
epsilon[0]=constvalue(%e_0),
x=6*m,y=-7*m,z=9*m],F(x,y,z)));
(%o13)  $\frac{1.615850173775426 \cdot 10^{-11} |C|}{m^2} \cdot \frac{m^2 C N}{s^2 A^2}$ 
```

In the final step we replace symbol “.” by “*”.

```
(%i14) (1.615850173775426*10^-11*C)/m^2 * (m^2*C*N)/(s^2*A^2);
(%o14)  $\frac{1.615850173775426 \cdot 10^{-11} C^2 N}{s^2 A^2}$ 
```

Simplifying units returns

```
(%i15) subst(A=C/s,%);
(%o15) 1.615850173775426 10^-11 N
```

Next page shows the respective DERIVE session followed by the TI-NspireCAS screens:

```

#1: [CaseMode := Sensitive, InputMode := Word]
#2: LOAD(C:\Program Files\TI Education\Derive 6\Math\PhysicalConstants.mth)
#3: 
$$q := e_-, \epsilon_0 := \frac{\frac{1}{36 \cdot \pi} \cdot 10^{-9} \cdot \text{coulomb}^2}{\text{newton} \cdot \text{meter}^2}$$

#4: 
$$Q_- := \left[ 4, -3, \frac{1}{2} \right] \cdot \text{coulomb}, R_- := \begin{bmatrix} 0 & 2 & 0 \\ -3 & 2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \cdot \text{meter}$$

#5:  $r_- := [x, y, z]$ 
#6: 
$$\phi := \frac{1}{4 \cdot \pi \cdot \epsilon_0} \cdot \sum_{j=1}^{\text{DIM}(Q_-)} \frac{Q_{-j}}{|r_- - R_{-j}|}$$

#7: [E_- := - GRAD( $\phi$ ), F_- := q·E_-]
#8: |SUBST(F_-, r_-, [6, -7, 9]·meter)|
#9:  $1.615890105 \cdot 10^{-11} \cdot |\text{newton}|$ 

```

TI-NspireCAS treatment is following. The gradient function is not provided by the system, so we use Michel Beaudin's library (presented in DNL#98) in order to apply his *grad* function.

Elementary charge q and electric constant (permittivity in a vacuum) ϵ_0 are provided among the "Constants" in the "Unit Conversions" which can be found in the Documents Toolbox.

© electron charge q and permittivity ϵ_0 are implemented as $_q$ and $_{\epsilon_0}$ (meter = $_m$).

$q := _q; \epsilon_0 := _{\epsilon_0}; m := _m$ 1.000000 · $_m$

$qq_- := [4 \ -3 \ 0.5] \cdot \text{coul}$ [4.000000 · $_coul$ -3.000000 · $_coul$ 0.500000 · $_coul$]

$rr_- := \begin{bmatrix} 0 & 2 & 0 \\ -3 & 2 & 0 \\ 0 & 0 & 0 \end{bmatrix} \cdot m$ [0 2.000000 · $_m$ 0 \\ -3.000000 · $_m$ 2.000000 · $_m$ 0 \\ 0 0 0]

$r_- := [x \ y \ z]$ [$x \ y \ z$]

$$\phi := \frac{1}{4 \cdot \pi \cdot \epsilon_0} \cdot \sum_{j=1}^{\text{dim}(qq_-)[2]} \left(\frac{qq_-[1,j]}{\text{norm}(r_- - rr_-[j])} \right)$$

-26962655362.1

$$\frac{-26962655362.1 \cdot \text{m} \cdot \text{V}}{\sqrt{x^2 + 6.000000 \cdot x \cdot \text{m} + y^2 - 4.000000 \cdot y \cdot \text{m} + z^2 + 13.000000 \cdot \text{m}^2} \cdot \sqrt{x^2 + y^2}}$$

$ee := \text{-kit ets mb} \backslash \text{grad}(\phi, [x \ y \ z])$

```

4 * pi * epsilon_0 * sum_{j=1}^n (norm(r_-rr_[j]))
-26962655362.1
sqrt(x^2+6.000000*x*_m+y^2-4.000000*y*_m+z^2+13.000000*_m^2) * _m*_V + sqrt(x^2+y^2)
ee_:=kit_ets_mb\grad(phi,[x y z])
-26962655362.1*(x+3.000000*_m)
(x^2+6.000000*x*_m+y^2-4.000000*y*_m+z^2+13.000000*_m^2)^1.500000*_m
ee_:=ee_|x=6*_m and y=-7*_m and z=9*_m
[25965674.5977*_N/_coul -66776938.0585*_N/_coul 70979162.7842*_N/_coul]
ff_:=q*ee_ [4.160160E-12*_N -1.069884E-11*_N 1.137212E-11*_N]
norm(ff_) 1.615850E-11*_N

```

II.4 Three point charges Q_1, Q_2, Q_3 are placed as shown in Fig II.4 .




Fig II.4

- Give the formula for the resulting electric field in a generic point $\vec{r}(x, y, z)$.
- Calculate all Cartesian components of the electric field on the x -axis.
- For identical charge ($Q_1 = Q_2 = Q_3 = Q$) and the distances $x_1 = x_2 = y_1 = y_3 = b$ calculate the parameter $\eta = \frac{y_2}{b}$ for which the field component E_y of the resultant field disappears at $x = 0$.
- Plot the curve $E_y(x)$ for the calculated parameter η .

Solution:

a) First we enter the given data:

```

(*i1) Q_:[Q1,Q2,Q3] $ r_: [x, y, z] $
(*i3) R_:matrix([x1,y1,0],[0,y2,0],[x3,y3,0]) $

```

According to the principle of superposition (page 16), the resultant field at the point \vec{r} can be expressed as

```
(%i4) E_:1/(4*pi*eps0)*
      sum(Q_[j]/sqrt((r_-R_[j]).(r_-R_[j]))^3*(r_-R_[j]),
          j,1,length(Q_))$
```

b) We calculate all Cartesian components of electric field along the x-axis:

```
(%i5) subst(y=0,z=0,E_);
```

$$\begin{aligned} & \frac{(x-x_3)Q_3}{(z^2+(y-y_3)^2+(x-x_3)^2)^{3/2}} + \frac{xQ_2}{(z^2+(y-y_2)^2+x^2)^{3/2}} + \frac{(x-x_1)Q_1}{(z^2+(y-y_1)^2+(x-x_1)^2)^{3/2}} \\ (%o5) & \left[\frac{\hspace{10em}}{4\pi\epsilon_0}, \right. \\ & \frac{(y-y_3)Q_3}{(z^2+(y-y_3)^2+(x-x_3)^2)^{3/2}} + \frac{(y-y_2)Q_2}{(z^2+(y-y_2)^2+x^2)^{3/2}} + \frac{(y-y_1)Q_1}{(z^2+(y-y_1)^2+(x-x_1)^2)^{3/2}} \\ & \left. \frac{\hspace{10em}}{4\pi\epsilon_0}, \right. \\ & \frac{zQ_3}{(z^2+(y-y_3)^2+(x-x_3)^2)^{3/2}} + \frac{zQ_2}{(z^2+(y-y_2)^2+x^2)^{3/2}} + \frac{zQ_1}{(z^2+(y-y_1)^2+(x-x_1)^2)^{3/2}} \\ & \left. \frac{\hspace{10em}}{4\pi\epsilon_0} \right] \end{aligned}$$

c) We extract the field component E_y :

```
(%i6) Ey:E_[2]$
```

At the origin of the reference system and for all identical charges and the distances $x_1 = x_2 = y_1 = y_3 = b$, $y_2 = -\eta b$ we have

```
(%i7) eq:ev(Ey,Q1=Q,Q2=Q,Q3=Q,
          x=0,y=0,z=0,
          x1=b,x2=b,x3=-b,y1=b,y2=-eta*b,y3=b);
```

$$(%o7) \frac{\frac{Q}{b|b|\eta|\eta|} - \frac{Q}{\sqrt{2}b|b|}}{4\pi\epsilon_0}$$

It is desired be equal to zero. To find the parameter η it remains to solve the above equation

```
(%i8) assume(b>0,eta>0)$
(%i9) solve(ev(eq)=0,eta);
(%o9) [\eta=-2^{1/4}, \eta=2^{1/4}]
```

Hence the desired parameter value is $\eta = 2^{1/4}$.


```
(%i10) eta:%[2];
(%o10) \eta=2^{1/4}
```

Then the explicit form of Ey is

```
(%i11) Ey:
      (y-y3) Q3      (y-y2) Q2      (y-y1) Q1
      ----- + ----- + -----
      (z^2+(y-y3)^2+(x-x3)^2)^(3/2)  (z^2+(y-y2)^2+x^2)^(3/2)  (z^2+(y-y1)^2+(x-x1)^2)^(3/2)
(%o11) -----
      4 pi eps0
```

(d) Now we plot the desired component $Ey(x)$ (for the calculated parameter η value) and the following remaining parameters values: $\epsilon_0 = 1$, $Q = 1$, $b = 1$ (needs loading library "draw").

```
(%i12) f(x):=ev(Ey,eps0=1,Q1=1,Q2=1,Q3=1,
              y=0,z=0,
              x1=1,x2=1,x3=-1,y1=1,y2=-2^(1/4),y3=1)$
(%i13) load(draw);
(%o13) C:/PROGRAMS/1/MAXIMA-1.2/share/maxima/5.23.2/share/draw/draw.lisp
(%i14) draw2d( title = "Ey(x)",
              explicit(f(x),x,-3,3));
```




The TI-Nspire treatment follows:

$$\begin{aligned}
& eex:=ee_{[1,1]}|y=0 \text{ and } z=0 \\
& \frac{qq1 \cdot (x-x1)}{4 \cdot eps0 \cdot \pi \cdot (x^2-2 \cdot x1 \cdot x+x1^2+y1^2)^{\frac{3}{2}}} + \frac{qq3 \cdot (x-x3)}{4 \cdot eps0 \cdot \pi \cdot (x^2-2 \cdot x3 \cdot x+x3^2+y3^2)^{\frac{3}{2}}} + \frac{qq2 \cdot x}{4 \cdot eps0 \cdot \pi \cdot (x^2+y2^2)^{\frac{3}{2}}} \\
& eey:=ee_{[1,2]}|y=0 \text{ and } z=0 \\
& \frac{-qq1 \cdot y1}{4 \cdot eps0 \cdot \pi \cdot (x^2-2 \cdot x1 \cdot x+x1^2+y1^2)^{\frac{3}{2}}} - \frac{qq3 \cdot y3}{4 \cdot eps0 \cdot \pi \cdot (x^2-2 \cdot x3 \cdot x+x3^2+y3^2)^{\frac{3}{2}}} - \frac{qq2 \cdot y2}{4 \cdot eps0 \cdot \pi \cdot (x^2+y2^2)^{\frac{3}{2}}} \\
& eez:=ee_{[1,3]}|y=0 \text{ and } z=0 \quad 0 \\
& eey|qq1=qq \text{ and } qq2=qq \text{ and } qq3=qq \text{ and } x1=b \text{ and } y1=b \text{ and } x3=-b \text{ and } y3=b \\
& \frac{-b \cdot qq}{4 \cdot eps0 \cdot \pi \cdot (x^2+2 \cdot b \cdot x+2 \cdot b^2)^{\frac{3}{2}}} - \frac{b \cdot qq}{4 \cdot eps0 \cdot \pi \cdot (x^2-2 \cdot b \cdot x+2 \cdot b^2)^{\frac{3}{2}}} - \frac{qq \cdot y2}{4 \cdot eps0 \cdot \pi \cdot (x^2+y2^2)^{\frac{3}{2}}} \\
& eey|qq1=qq \text{ and } qq2=qq \text{ and } qq3=qq \text{ and } x1=b \text{ and } y1=b \text{ and } x3=-b \text{ and } y3=b \text{ and } x=0 \text{ and } y2=-\eta \cdot b \\
& \quad \quad \quad qq \cdot \left(\frac{1}{4 \cdot \eta \cdot |\eta| \cdot \pi} - \frac{\sqrt{2}}{8 \cdot \pi} \right)
\end{aligned}$$

It is not necessary to add any restrictions for the variables for solving the equation.

$$\begin{aligned}
& \text{solve} \left(\frac{qq \cdot \left(\frac{1}{4 \cdot \eta \cdot |\eta| \cdot \pi} - \frac{\sqrt{2}}{8 \cdot \pi} \right)}{b \cdot |b| \cdot eps0} = 0, \eta \right) \\
& \quad \quad \quad b \cdot |b| \cdot eps0 \\
& \quad \quad \quad \frac{1}{\eta=2^{\frac{1}{4}} \text{ or } qq=0} \\
& eey|eps0=1 \text{ and } qq1=1 \text{ and } qq2=1 \text{ and } qq3=1 \text{ and } x1=1 \text{ and } y1=1 \text{ and } x3=-1 \text{ and } y3=1 \text{ and } y2=-2^{\frac{1}{4}} \\
& \quad \quad \quad \frac{-1}{4 \cdot \pi \cdot (x^2+2 \cdot x+2)^{\frac{3}{2}}} - \frac{1}{4 \cdot \pi \cdot (x^2-2 \cdot x+2)^{\frac{3}{2}}} + \frac{1}{2 \cdot 2^{\frac{1}{4}} \cdot 4 \cdot \pi \cdot (x^2+\sqrt{2})^{\frac{3}{2}}} \\
& f1(x):= \frac{-1}{4 \cdot \pi \cdot (x^2+2 \cdot x+2)^{\frac{3}{2}}} - \frac{1}{4 \cdot \pi \cdot (x^2-2 \cdot x+2)^{\frac{3}{2}}} + \frac{1}{2 \cdot 2^{\frac{1}{4}} \cdot 4 \cdot \pi \cdot (x^2+\sqrt{2})^{\frac{3}{2}}} \quad \text{Done}
\end{aligned}$$

Now we can plot $f1(x)$.



II.5 Equal charges of value Q are placed in the vertices of a regular hexagon with sides of length a .




Fig II.5

Calculate:

- the potential of the field in the centre of the hexagon,
- the electric field and potential in the centre of one given side of your choice.

Solution:

Remember, the potential of the electric field at the point \vec{r} resulting from a charge Q_j placed in a point \vec{R}_j is described by the relation

$$\Phi_j = \frac{1}{4\pi\epsilon_0} \frac{Q_j}{|\vec{r} - \vec{R}_j|},$$

In the problem discussed $Q_j = Q$, ($j=1, 2, \dots, 6$). Hence, the resultant potential can be written in the form

$$\Phi = \frac{1}{4\pi\epsilon_0} \sum_{j=1}^6 \frac{Q_j}{|\vec{r} - \vec{R}_j|} = \frac{Q}{4\pi\epsilon_0} \sum_{j=1}^6 \frac{1}{|\vec{r} - \vec{R}_j|}.$$

The components of the position vectors \vec{R}_j , taking the centre of the hexagon as the origin of the coordinate system, are given by:

$$x_j = a \cos\left(\frac{\pi}{3}j\right), y_j = a \sin\left(\frac{\pi}{3}j\right), z_j = 0, \quad (j=1,2,\dots,6)$$

We enter the formulae for the position vectors of the charges, the position vector of the observed point and the resultant potential according to the equation from above:

```
(%i1) R_(j):=[a*cos(%pi/3*j),a*sin(%pi/3*j),0]$
(%i2) assume(a>0)$
(%i3) r_: [x,y,z]$
(%i4) Phi:Q/(4*pi*eps0)*sum(1/sqrt((r_-R_(j)).(r_-R_(j))),j,1,6)$
```

a) Then the potential in the centre of the hexagon is

```
(%i5) Phi,x=0,y=0,z=0;
(%o5) 
$$\frac{3Q}{2\pi a \epsilon_0}$$

```

b) In the centre of any given side the potential is

```
(%i6) Phi,x=0,y=a/2*sqrt(3),z=0;
(%o6) 
$$\frac{\left(\frac{4}{\sqrt{13}a} + \frac{4}{\sqrt{7}a} + \frac{4}{a}\right)Q}{4\pi\epsilon_0}$$

```

To calculate the field we enter the well known relation between field and the potential

$$\vec{E} = -\text{grad}(\Phi).$$

and then we calculate the field in the centre of one side of the hexagon (let's say in the point

$$\left(x=0, y=\frac{a}{2}\sqrt{3}, z=0\right).$$

```
(%i8) E_-grad(Phi)$
(%i9) E_:ev(express(E_),diff)$
(%i10) E_,x=0,y=a/2*sqrt(3),z=0;
(%o10) 
$$\left[0, \frac{\left(\frac{16\sqrt{3}}{13^{3/2}a^2} - \frac{8\sqrt{3}}{7^{3/2}a^2}\right)Q}{4\pi\epsilon_0}, 0\right]$$

(%i11) radcan(%);
(%o11) 
$$\left[0, \frac{(2\sqrt{3}13^{3/2} + 4\sqrt{3}7^{3/2})Q}{7^{3/2}13^{3/2}\pi a^2 \epsilon_0}, 0\right]$$

```

DERIVE and TI-NspireCAS treatments are very similar.

Exercise: Compute the magnitude of the electric field and potential in the centres of all remaining sides of the hexagon and compare the obtained results.

II.6 There are five charges Q_1, Q_2, Q_3, Q_4 and Q_0 located as shown in Fig. II.6.




Fig. II.6

- Find the relation between charges Q_1, Q_2, Q_3, Q_4 for which the resulting acting on the charge Q_0 disappears?
- Plot the equipotential curves and the 3D-plot of the potential.

Solution:

One of the solutions, namely: $Q_1 = Q_2 = Q_3 = Q_4$ (full symmetry) is obvious. We try to find the remaining solutions.

- In the problem we have

```
(%i1) R_:matrix([a,a,0],[-a,a,0],[-a,-a,0],[a,-a,0])/2$
      assume(a>0)$
      r_: [x,y,z]$  Q_: [Q1,Q2,Q3,Q4]$
```

To find the remaining solutions we take advantage of relations met in the previous task. The potential of the electric field at the point \vec{r} resulting from a charge Q_j placed in a point \vec{R}_j is described by the relation

```
(%i5) Phi:1/(4*pi*eps0)*
      sum(Q_[j]/sqrt((r_-R_[j]).(r_-R_[j])),j,1,4)$
```

Let us apply the connection between potential and field strength and calculate the field strength at the location point of charge Q_0 .

```
(%i6) load(vect)$ E_-:grad(Phi)$
(%i8) E_:ev(express(E_),diff)$
      E0_:ev(E_,x=0,y=0,z=0)$
(%i10) E0x:factor(E0_[1]);E0y:factor(E0_[2]);E0z:factor(E0_[3]);
(%o10) 
$$\frac{Q_4 - Q_3 - Q_2 + Q_1}{2^{3/2} \pi a^2 \epsilon_0}$$

(%o11) 
$$\frac{Q_4 + Q_3 - Q_2 - Q_1}{2^{3/2} \pi a^2 \epsilon_0}$$

(%o12) 0
```

The electric field vector disappears in the desired point if every its component is zero. This implies that we have to solve two equations. We try to solve this equation system e.g. for the unknowns Q_1 and Q_2 .

```
(%i13) solve([E0x,E0y],[Q1,Q2]);
(%o13) [[Q1=Q3, Q2=Q4]]
```

From the received solution we can conclude that the resultant power acting on charge Q_0 disappears when conditions $Q_1 = Q_3$ and $Q_2 = Q_4$ are fulfilled.

Of course, if every component of any vector is equal zero then its value (length of the vector) is equal to zero. We are getting the task to the solution of only one equation

$$\left(\vec{E}\vec{E}\right)_{\vec{r}=[0,0,0]} = 0.$$

We try to solve this equation for one unknown e.g. for Q_1 :

```
(%i14) solve(E0_.E0_,Q1); solve(E0_.E0_,Q2);
(%o14) [Q1=-%i Q4+Q3+%i Q2, Q1=%i Q4+Q3-%i Q2]
(%o15) [Q2=Q4-%i Q3+%i Q1, Q2=Q4+%i Q3-%i Q1]
```

As charges are real we deduce from %o14 that $Q_2 = Q_4$ and further $Q_1 = Q_3$.

Of course, the same relations between charges are being received by solving the equation with respect to any other charge e.g. Q_2 (see above %o15).

We can obtain solution of the problem in a single step applying the solve command:


```
(%i16) solve(E0_,[Q1,Q2]);
solve: dependent equations eliminated: (3)
(%o16) [[Q1=Q3, Q2=Q4]]
```

b) Let's produce the plots:

```
(%i17) eps0:1$ a:1$ Q1:1$ Q2:-1$ Q3:1$ Q4:Q2$
(%i23) Phiz0:ev(Phi,z=0)$
(%i24) load(draw)$
```


We generate the 3D-plot of the potential.

```
(%i27) wxplot3d(Phiz0,[x,-2,2],[y,-2,2],[color,magenta]);
```




```
(%o27)
```

This is another form of the plot:





The equipotential curves in the xy -plane are plotted:

```
(%i26) draw3d(title="Contours of the Potential",
  explicit(Phiz0,x,-1,1,y,-1,1),
  contour_levels = 25,
  contour = map,
  surface_hide = true);
```



The two plots below show the DERIVE plots (after similar calculations):










The TI-NspireCAS 3D-plot of the potential looks pretty good (left hand side), but unfortunately it is not possible to produce its contour plot.

What we can do to get an impression of the contour lines, is tracing the surface.

See below the trace for $z = 0.08$ (left) and $z = 0.16$ (right).



With the symbolic pocket calculator Voyage 200 we can plot the surface and its contour plot as well. The screen shots right and below are demonstrating this. It is understandable that we cannot expect high quality plots like those done on a PC but for a hand held product the plots are pretty good.



II.7 Four identical charges q are placed in the vertices of a square, whose sides are of length a . (Fig. II.7)




Fig. II.7

- a) Which charge Q should be placed in the centre of the square, in order for the entire system to be in equilibrium?
 - b) Assuming the charges q are stationary, write expressions for the force acting on charge Q after moving it from the centre by a distance b in direction:
 - ba) of one of the vertices,
 - bb) of the centre of one of the sides,
 - bc) perpendicular to the plane of the square.
- Give approximations for these expressions in case of $b \ll a$.

Solution:

- a) According to *Coulomb's law* the force acting on a charge q_j placed in \vec{R}_j , due to charge q_i in \vec{R}_i is given by

$$\vec{F}_{i,j} = \frac{1}{4\pi\epsilon_0} \frac{q_i q_j}{|\vec{R}_j - \vec{R}_i|^3} (\vec{R}_j - \vec{R}_i).$$

```
(%i1) q_:[q,q,q,q]$
(%i2) R_:1/2*matrix([-a,-a,0],[a,-a,0],[a,a,0],[-a,a,0])$
and Coulomb's Law:
(%i3) F_(i,j):=if i=j then 0 else
      1/(4*pi*eps0)*(q_[i]*q_[j])*(R_[j]-R_[i])/
      sqrt((R_[j]-R_[i]).(R_[j]-R_[i]))^3;
(%o3) F_(i,j):=if i=j then 0 else
      1/(4*pi*eps0)*(q_[i]*q_[j])*(R_[j]-R_[i])/
      sqrt((R_[j]-R_[i]).(R_[j]-R_[i]))^3
```

The total force acting on a charge q_j is the vector sum of all forces due to the charges placed in all remaining vertices of the square and the force due to charge Q (with position vector $\vec{r} = \mathbf{r}_j$).

```
(%i4) r_: [0,0,0]$
(%i5) Ftotal_(j):=sum(F_(i,j),i,1,length(q_))+
      1/(4*pi*eps0)*(Q*q_[j])*(R_[j]-r_)/
      sqrt((R_[j]-r_).(R_[j]-r_))^3;
(%o5) Ftotal_(j):=
      sum_{i=1}^{length(q_)} F_(i,j) +
      1/(4*pi*eps0)*(Q*q_[j])*(R_[j]-r_)/
      sqrt((R_[j]-r_).(R_[j]-r_))^3
```

The system will be in equilibrium when the resultant force acting on one of the charges q from the other charges equals zero. First we will find the required charge Q for $j = 1$.

```
(%i6) Ftotal(j):=sqrt(Ftotal_(j).Ftotal_(j))$
(for j = 1,...,4)
(%i7) Ftotal(1);
(%o7)  $\sqrt{2} \left| \frac{qQ}{2^{3/2} \pi a |a| \epsilon_{ps0}} + \frac{q^2}{2^{7/2} \pi a |a| \epsilon_{ps0}} + \frac{q^2}{4 \pi a |a| \epsilon_{ps0}} \right|$ 
(%i8) factor(%);
(%o8)  $\frac{|q| |4Q + (2^{3/2} + 1)q|}{8 \pi a^2 |\epsilon_{ps0}|}$ 
(%i9) solve(%o8=0,Q);
(%o9)  $[Q = -\frac{(2^{3/2} + 1)q}{4}]$ 
```


The total force should be zero for all indices j . Let's verify the obtained solution:

```
(%i10) Q:-(2^(3/2)+1)/4*q$
(%i11) ratsimp([Ftotal(1),Ftotal(2),Ftotal(3),Ftotal(4)]);
(%o11) [0,0,0,0]
```

b) Now we will calculate the force \overline{FQ} arising on charge Q due to its displacement. Q is placed at position $\vec{r}(x,y,z)$.

```
(%i12) r_: [x,y,z]$
(%i13) FQ_:Q/(4*pi*eps0)*sum((q_[j])*(r_-R_[j])/
sqrt((R_[j]-r_) . (R_[j]-r_))^3,j,1,length(q_))$
```

ba) The resulting force acting on charge Q




can be entered in the following form (%i14). For small displacements $b \ll a$ we will introduce the parameter δ ($b = \delta a$) and then perform the Taylor expansion of force $\overline{FQ} = FQ_+$ up to the first order term.

```
(%i14) ex:ev(FQ_,x=b/sqrt(2),y=b/sqrt(2),z=0)$
(%i15) assume(a>0,eps0>0)$
(%i16) taylor(ev(ex,b=delta*a),delta,0,1);
(%o16) /T/  $[\frac{(2q^2\sqrt{2}+q^2)\delta}{4a^2\pi\epsilon_{ps0}} + \dots, \frac{(2q^2\sqrt{2}+q^2)\delta}{4a^2\pi\epsilon_{ps0}} + \dots, 0 + \dots]$ 
```

We see that the calculated force is repulsive. Let's calculate the magnitude of this force, too:

```
(%i17) sqrt(%.);
taylor: assumed to be zero: 0+...
(%o17)/T/  $\frac{\sqrt{4\sqrt{2}+9} q^2 \sqrt{2} \delta}{4 \text{eps}0 a^2 \pi} + \dots$ 
```

bb) Now charge Q moves towards the centre of a side of the square:




For a small displacement we get:

```
(%i18) ex:ev(FQ_,x=a*delta,y=0,z=0)$
(%i19) taylor(ex,delta,0,1);
(%o19)/T/  $[\frac{(q^2\sqrt{2}+4q^2)\delta}{4a^2\pi\text{eps}0} + \dots, 0 + \dots, 0 + \dots]$ 
```

We see that the calculated force is repulsive again.

bc) In the last case we have a displacement in direction perpendicular to the plane of the square (see next figure).



We perform the same procedure as above and we find out that now the resulting force is attractive.

```
(%i20) ex:ev(FQ_,x=0,y=0,z=a*delta)$
(%i21) taylor(ex,delta,0,1);
(%o21)/T/  $[0 + \dots, 0 + \dots, -\frac{(\sqrt{2}+4)q^2\delta}{2\pi a^2\text{eps}0} + \dots]$ 
```

Resuming we see (results %o16, %o19 and %o21) that in point (0,0,0) the equilibrium state of charge Q is an unstable one.

DERIVE and TI-NspireCAS treatment as well are easy to perform and deliver the same results.

II.8 Two charges of absolute value Q and opposite signs are placed at a distance of d from each other (an electric dipole). The origin of the coordinate system is taken to be the centre of the dipole. The configuration of the dipole is shown in Fig. II.8




Fig. II.8

- Calculate the potential of such a dipole in a generic point $P(x,y,z)$.
- Plot the equipotential lines for the dipole.
- How may one estimate the potential for the case that distance d is much smaller than distance r of point $P(x, y, z)$ from the centre of the dipole?
- Estimate the electric field for the case that distance d is much smaller than distance r of point $P(x,y,z)$ from the centre of the dipole.

Solution:

- We start with the potential in point \vec{r} , due to a single charge Q placed in point P given by its position vector \vec{R} .

The resultant potential is the sum of the potentials generated by the individual charges, therefore we get:

```
(%i1) r_: [x, y, z]$
(%i2) Phi(Q, r_, R_) := (1 / (4 * %pi * eps0)) * (Q / sqrt((r_ - R_) . (r_ - R_)))$
(%i3) R1_: [0, 0, -d/2]$ R2_: [0, 0, d/2]$
(%i5) Phidipol: Phi(-Q, r_, R1_) + Phi(Q, r_, R2_);
(%o5) 
$$\frac{Q}{4 \pi \text{eps}0 \sqrt{\left(z - \frac{d}{2}\right)^2 + y^2 + x^2}} - \frac{Q}{4 \pi \text{eps}0 \sqrt{\left(z + \frac{d}{2}\right)^2 + y^2 + x^2}}$$

```

In the symmetry plane of the dipole the potential is zero.

```
(%i6) ev(Phidipol, z=0);
(%o6) 0
```

Along the dipole symmetry axis we get:


```
(%i7) ev(Phidipol, x=0, y=0);
(%o7) 
$$\frac{Q}{4 \pi \text{eps}0 \left|z - \frac{d}{2}\right|} - \frac{Q}{4 \pi \text{eps}0 \left|z + \frac{d}{2}\right|}$$

```

b) Let's plot the equipotential curves in the xz -plane (i.e. for $y = 0$):

```
(%i8) Phidipol0:ev(Phidipol,Q=1,d=1,eps0=1,y=0);
(%o8) 
$$\frac{1}{4\pi\sqrt{\left(z-\frac{1}{2}\right)^2+x^2}} - \frac{1}{4\pi\sqrt{\left(z+\frac{1}{2}\right)^2+x^2}}$$

--> load(draw)$
--> draw3d(title="Equipotential Lines",
explicit(Phidipol0,x,-0.4,0.4,z,-1,1),
contour_levels=25,
contour=map,
surface_hide=true);
```



c) In order to estimate the potential of the dipole we use the parameter $\delta = \frac{d}{r}$ describing the ratio of the length of the dipole to the distance of the observed point from the centre of the dipole and then calculate the Taylor expansion of the potential with respect to the parameter δ preserving only the first order term of the expansion. Then we will return to the previous variables.

```
(%i9) r:sqrt(r_.r_)$
(%i10) ex:ev(Phidipol,d=delta*r)$
(%i11) Phiapprox:factor(taylor(ex,delta,0,1));
(%o11) 
$$\frac{\delta z Q}{4\pi\epsilon_0(z^2+y^2+x^2)}$$

(%i12) Phiapprox:ev(Phiapprox,delta=d/r);
(%o12) 
$$\frac{d z Q}{4\pi\epsilon_0(z^2+y^2+x^2)^{3/2}}$$

```

Along the symmetry axis of the dipole we get:

```
(%i13) ev(Phiapprox,x=0,y=0);
(%o13) 
$$\frac{dQ}{4\pi\epsilon_0 z|z|}$$

```

The approximated potential (result o%12) can be written in a compact form

$$\frac{1}{4\pi\epsilon_0} \frac{\vec{p} \cdot \vec{r}}{r^3}$$

where \vec{p} denotes the dipole moment $\vec{p} = [0, 0, Qd]$. Below is the verification of this compact form.

```
(%i14) p_: [0,0,Q*d]$
(%i15) Phiapprox-(1/(4*pi*eps0))*(p_.r_)/r^3;
(%o15) 0
```

d) For the field we enter the known relation

```
(%i18) load(vect)$
(%i19) E_-grad(Phidipol)$
(%i20) E_:ev(express(E_),diff)$
```

The approximated field can be calculated in a similar way as the approximated potential:

```
(%i21) E_:ev(E_,d=delta*r)$
(%i22) Eapprox_:taylor(E_,delta,0,1)$
(%i23) Eapprox_:ev(Eapprox_,delta=d/r)$
```

The approximated field (hidden in %i23) can also be written in compact form

$$\frac{1}{4\pi\epsilon_0} \frac{3\vec{p} \cdot \vec{r}}{r^5} \vec{r} - \frac{\vec{p}}{r^3}$$

See the verification:

```
(%i24) factor(Eapprox_-1/(4*pi*eps0)*(3*(p_.r_)*r_/r^5-p_/r^3));
(%o24) [0,0,0]
```


It is not difficult to produce the contour plot with DERIVE, all calculations and TI-NspireCAS treatment as well are easy to perform and deliver the same results. We show some DERIVE lines followed by the contour plot (plot of expression #7)

```
#4: 
$$\Phi_{dipol} := \frac{Q \cdot (\sqrt{(4 \cdot x^2 + 4 \cdot y^2 + (2 \cdot z + d)^2)} - \sqrt{(4 \cdot x^2 + 4 \cdot y^2 + (2 \cdot z - d)^2)})}{2 \cdot \pi \cdot \epsilon_0 \cdot \sqrt{(4 \cdot x^2 + 4 \cdot y^2 + (2 \cdot z + d)^2)} \cdot \sqrt{(4 \cdot x^2 + 4 \cdot y^2 + (2 \cdot z - d)^2)}}$$





#7: VECTOR(SUBST(Phi_dipol, [epsilon_0, Q, d, y], [1, 1, 1, 0])) =  $\frac{n}{100}, n, -5, 5$ 
```

#18:

$$E_{\text{approx}} = \frac{1}{4 \cdot \pi \cdot \epsilon_0} \cdot \left(\frac{3 \cdot (p_- \cdot r_-)}{r^5} \cdot r_- - \frac{p_-}{r^3} \right) = [0, 0, 0]$$



With TI-Nspire we proceed like in Example II.6: we plot the surface and then trace in y -direction which is upwards in our case (left: $y = 0.2$, right: $y = -0.04$).



Exercise: There are three point charges located on a line (linear quadrupole). Plot the equipotential lines and produce a 3D plot of the potential of this quadrupole with charge $-q$ at $x = -a$, the second charge $-q$ at $x = a$ and the third charge $2q$ located at the origin of the reference system.



II.9 Point charge Q is located at a distance b from the centre of a grounded conducting sphere of radius R ($b < R$). Find the position and magnitude of an image charge q which generates a zero potential on the spherical surface. Plot the equipotential lines.

Solution:

The potential at a distance r from point charge Q can be expressed as

$$\Phi(Q, r) = \frac{1}{4\pi\epsilon_0} \frac{Q}{r}.$$

By symmetry, the image charge must lie on the straight line connecting Q and the centre of the sphere.




Fig. II.9

The potential must be zero on the surface of the sphere. In particular it must be zero in the nearest and farthest points on the equator of the sphere which contains Q . Hence we build two equations and solve the system for a and q .

```
(%i1) Phi(Q, r) := 1 / (4 * pi * eps0) * (Q / r);
(%o1) Phi(Q, r) := 1 / (4 * pi * eps0) * Q / r
(%i2) eq1: Phi(Q, b - R) + Phi(q, R - a) = 0$
(%i3) eq2: Phi(Q, b + R) + Phi(q, R + a) = 0$
(%i4) solve([eq1, eq2], [q, a]);
(%o4) [[q = -Q * R / b, a = R^2 / b]]
```

From %o4 we deduce magnitude q and distance a of the image charge:


```
(%i5) q: -(Q * R) / b$    a: R^2 / b$
```

Then the potential is

```
(%i7) potential: Phi(Q, sqrt(x^2 + y^2 + (b - z)^2)) +
Phi(q, sqrt(x^2 + y^2 + (a - z)^2));
(%o7) Q / (4 * pi * eps0 * sqrt((b - z)^2 + y^2 + x^2)) -
(Q * R) / (4 * pi * b * eps0 * sqrt((R^2 / b - z)^2 + y^2 + x^2))
```


The equipotential lines lie in the xz -plane. They are presented on the next page ($\epsilon_0 = 1$, $Q = 1$, $b = 5$, $R = 3$).

```
(%i8) load(draw)$
(%i10) py0:ev(potential,y=0,eps0=1,Q=1,b=5,R=3)$
(%i18) contour_plot(py0,[x,-0.5,0.5],
                   [z,-0.2,6],
                   [legend,false],
                   [gnuplot_preamble,"set cntrparam levels 35"])$
```



There is no problem calculating and plotting with DERIVE:

$$\text{VECTOR} \left(\frac{1}{4 \cdot \pi \cdot \sqrt{(5-z)^2 + 0^2 + x^2}} - \frac{3}{4 \cdot \pi \cdot 5 \cdot \sqrt{\left(\left(\frac{3}{5} - z\right)^2 + 0^2 + x^2\right)}} = k, k, -1, 1, \frac{1}{20} \right)$$



The following problems deal with systems of continuous charge distributions.

II.10 Calculate the electrical field potential and the field intensity vector for the field created by a thin rod of length l with uniform charge density.

Solution: The configuration of the system is shown in Fig. II.10.




Fig. II.10

The field caused at point $P(x,y)$ due to the charge on an infinitesimally small segment of the rod dx' is given by

$$d\Phi = \frac{1}{4\pi\epsilon_0} \frac{\lambda dx'}{r},$$

where σ denotes the line charge density, r is the distance of the observed point P from the element dx' .

$$r = \sqrt{y^2 + (x-x')^2}.$$

The resultant potential is expressed by the following integral:

$$\Phi = \frac{1}{4\pi\epsilon_0} \int_0^l \frac{\lambda dx'}{r}.$$

Then the field intensity is the gradient of the field potential:

$$\vec{E} = -\text{grad}(\Phi).$$

We transfer these expressions into "Maxima language" and then calculate the potential Φ :

```
(%i1) r:sqrt((x-xp)^2+y^2)$
      assume(l>0,y>0,x>0)$
(%i3) Phi:1/(4*pi*eps0)*%lambda*integrate(1/r,xp,0,l);
      lambda*asin(x/y)-asin((x-1)/y)
(%o3) -----
      4 pi eps0
```

At the symmetry axis of the rod ($x = l/2$) we get

```
(%i4) Phi,x=l/2;
      lambda*asin(1/(2*y))
(%o4) -----
      2 pi eps0
```

Let' now find the field intensity:

```
(%i5) load(vect)$
      E_-:grad(Phi)$
(%i7) E_-:express(E_-),diff;
      λ ⎛ 1 1 ⎞ λ ⎛ x-1 x ⎞
      ⎜ √(x²+1) y √((x-1)²+1) y ⎟ ⎜ √((x-1)²+1) y² √(x²+1) y² ⎟
(%o7) [ -————, -————, 0 ]
      4 π eps0 4 π eps0
```

For the axis of symmetry Maxima gives:

```
(%i8) E_-,x=1/2;
(%o8) [ 0, —————, 0 ]
      λ l
      4 π eps0 √(l²/4 + 1) y²
```

For an infinite long rod the field is inverse proportional to the distance from the rod:

```
(%i9) limit(%o8,l,inf);
(%o9) [ 0, —————, 0 ]
      λ
      2 π eps0 y
```

It is interesting to perform the calculation from above using another CAS, e.g. DERIVE:

#4:
$$\text{SUBST}\left(\phi, x, \frac{1}{2}\right) = - \frac{\lambda \cdot \text{LN}\left(\frac{(\sqrt{(4 \cdot y^2 + 1)} - 1)^2}{4 \cdot y^2}\right)}{4 \cdot \pi \cdot \epsilon_0}$$

#5:
$$\text{SUBST}\left(E_-, x, \frac{1}{2}\right) = \left[0, \frac{1 \cdot \lambda}{2 \cdot \pi \cdot \epsilon_0 \cdot y \cdot \sqrt{(4 \cdot y^2 + 1)}}, 0 \right]$$

#6:
$$\lim_{l \rightarrow \infty} \text{SUBST}\left(E_-, x, \frac{1}{2}\right) = \left[0, \frac{\lambda}{2 \cdot \pi \cdot \epsilon_0 \cdot y}, 0 \right]$$

#7:
$$\phi = \frac{\lambda \cdot \text{LN}\left(\frac{(\sqrt{(x^2 + y^2)} + x) \cdot (\sqrt{(x^2 - 2 \cdot l \cdot x + y^2 + 1)} - x + 1)}{y^2}\right)}{4 \cdot \pi \cdot \epsilon_0}$$

%o4 (Maxima) should be identical with #4 (DERIVE):

$$\frac{\lambda \cdot \text{ASINH}\left(\frac{1}{2 \cdot y}\right)}{2 \cdot \pi \cdot \epsilon_0} - \frac{\lambda \cdot \text{LN}\left(\frac{(\sqrt{(4 \cdot y^2 + 1)} - 1)^2}{4 \cdot y^2}\right)}{4 \cdot \pi \cdot \epsilon_0} = 0$$

Identity of %o8 and #5 can be checked easily manually and %o9 ≡ #6. What about Φ and its AREA-functions?

$$\text{LN} \left[\frac{(\sqrt{(x^2 + y^2) + x}) \cdot (\sqrt{(x^2 - 2 \cdot l \cdot x + y^2 + l^2)} - x + l)}{y^2} \right] - \left(\text{ASINH} \left(\frac{x}{y} \right) - \text{ASINH} \left(\frac{x - l}{y} \right) \right) = 0$$

The expression for the potential is rather complicated in DERIVE, the expression for the field intensity is complicated in both systems. However, field intensity can take a much simpler form introducing angles α_1 and α_2 (see Fig. II.10a).




Fig. II.10a

We can read off: $\tan(\alpha_1) = -\frac{d}{b}$ and $\tan(\alpha_2) = \frac{d}{b-l} \rightarrow b = -\frac{d}{\tan(\alpha_1)}, l = b - \frac{d}{\tan(\alpha_2)}$.

#13: $\left[b := -\frac{d}{\text{TAN}(\alpha_1)}, l := b - \frac{d}{\text{TAN}(\alpha_2)} \right]$

#14: $\text{SUBST}(\Phi, [x, y], [b, d]) = \frac{\lambda \cdot \text{LN} \left(\text{TAN} \left(\frac{\alpha_1}{2} \right) \cdot \text{TAN} \left(\frac{\alpha_2}{2} \right) \right)}{4 \cdot \pi \cdot \epsilon_0}$

#15: $\text{SUBST}(E_-, [x, y], [b, d]) = \left[\frac{\lambda \cdot \text{SIN}(\alpha_2)}{4 \cdot \pi \cdot d \cdot \epsilon_0} - \frac{\lambda \cdot \text{SIN}(\alpha_1)}{4 \cdot \pi \cdot d \cdot \epsilon_0}, -\frac{\lambda \cdot \text{COS}(\alpha_1)}{4 \cdot \pi \cdot d \cdot \epsilon_0} - \frac{\lambda \cdot \text{COS}(\alpha_2)}{4 \cdot \pi \cdot d \cdot \epsilon_0}, 0 \right]$

#16: $\text{SUBST}(\text{SUBST}(E_-, [x, y], [b, d]), [\alpha_1, \alpha_2], [\pi, \pi]) = \left[0, \frac{\lambda}{2 \cdot \pi \cdot d \cdot \epsilon_0}, 0 \right]$

In this case it turns out that the Maxima calculation is more complicated:

```
(%i15) EE_:E_, x=b, y=d;
      lambda * ( 1 / (sqrt(1/tan(alpha_1)^2 + 1) * d) - 1 / (sqrt(1/tan(alpha_2)^2 + 1) * d) )
(%o15) [ -----, -----, 0
      4 * pi * eps0          4 * pi * eps0
]
(%i16) trigsimp(%);
(%o16) [ (|sin(alpha_2)| - |sin(alpha_1)|) * lambda / (4 * pi * d * eps0), (sin(alpha_1) * cos(alpha_2) * |sin(alpha_2)| + cos(alpha_1) * sin(alpha_2) * |sin(alpha_1)|) * lambda / (4 * pi * sin(alpha_1) * sin(alpha_2) * d * eps0), 0 ]
```

But finally we also reach our goal!

II.11 Calculate the force of the mutual interaction of two uniformly charged rods of length L_1 and L_2 lying on the same line and of charge Q_1 and Q_2 respectively. The distance between the rods is d . (Parameters for the plot: $\epsilon_0 = 1$, $Q_1 = 1$, $Q_2 = 2$, $L_1 = 0.5$, $L_2 = 1.5$)

Solution:

We use *Coulomb's* law:




Fig. II.11

The force of interaction between two charged segments of lengths dx_1 and dx_2 with coordinates x_1 and x_2 (starting points of the segments) is given by

$$dF = \frac{1}{4\pi\epsilon_0} \frac{\lambda_1 \lambda_2}{(x_1 - x_2)^2} dx_1 dx_2$$

where λ_1, λ_2 are the charge densities of the rods ($\lambda_1 = \frac{Q_1}{L_1}$, $\lambda_2 = \frac{Q_2}{L_2}$).

The resultant force is obtained by integration

$$F = \frac{1}{4\pi\epsilon_0} \int_0^{L_1} \left(\int_{L_1+d}^{L_1+d+L_2} \frac{\lambda_1 \lambda_2}{(x_2 - x_1)^2} dx_2 \right) dx_1 = \frac{Q_1 Q_2}{4\pi\epsilon_0 L_1 L_2} \int_0^{L_1} \left(\int_{L_1+d}^{L_1+d+L_2} \frac{1}{(x_2 - x_1)^2} dx_2 \right) dx_1.$$

We enter the expressions from above and proceed with the integration:


```
(%i1) %lambda[1]:Q[1]/L[1]$      %lambda[2]:Q[2]/L[2]$
%lambda give the greek character λ.
(%i3) dF:%lambda[1]*%lambda[2]/(4*pi*%epsilon[0])/(x[2]-x[1])^2;
(%o3)
      Q1 Q2
-----
4 π ε0 L1 (x2 - x1)² L2
.
(%i4) assume(L[1]>0,L[2]>0,d>0)$
(%i5) F:integrate(integrate(dF,x[1],0,L[1]),x[2],L[1]+d,L[1]+d+L[2]);
Is x2-L1 positive, negative or zero?p;
(%o5)
      Q1 Q2 (-log(d+L2+L1)+log(d+L2)+log(d+L1)-log(d))
-----
      4 π ε0 L1 L2
```

%o5 can be simplified to the compact form:

```
(%i6) logcontract(%);
(%o6)
      Q1 Q2 log
-----
      4 π ε0 L1 L2
      (d + L2 + L1)
      (d + L1) (d + L2)
```

Now we can assume parameter values and then plot force F vs distance d :

```
(%i7) ev(%o5,%epsilon[0]=1, Q[1]=1,Q[2]=2,L[1]=1/2,L[2]=3/2);
      2 \left( -\log(d+2)+\log\left(d+\frac{3}{2}\right)+\log\left(d+\frac{1}{2}\right)-\log(d) \right)
(%o7) -----
      3 \pi
(%i8) load(draw)$
      draw2d(title="F(d)",
      explicit(%o7,d,2,5));
```



Here are the TI-NspireCAS screens:

$$\lambda_1 := \frac{q_1}{l_1}; \lambda_2 := \frac{q_2}{l_2} \rightarrow \frac{q_2}{l_2}$$


$$df := \frac{\lambda_1 \cdot \lambda_2}{4 \cdot \pi \cdot \epsilon_0 \cdot (x_2 - x_1)^2} \rightarrow \frac{q_1 \cdot q_2}{4 \cdot l_1 \cdot l_2 \cdot (x_1 - x_2)^2 \cdot \epsilon_0 \cdot \pi}$$

$$\int_0^{l_1} \int_0^{l_1+d+l_2} \frac{df \, dx_1 \, dx_2 | l_1 > 0 \text{ and } l_2 > 0 \text{ and } d > 0}{l_1+d} \rightarrow \frac{-\ln\left(\frac{d \cdot (d+l_1+l_2)}{(d+l_2) \cdot (d+l_1)}\right) \cdot q_1 \cdot q_2}{4 \cdot l_1 \cdot l_2 \cdot \epsilon_0 \cdot \pi}$$

$$\frac{-\ln\left(\frac{d \cdot (d+l_1+l_2)}{(d+l_2) \cdot (d+l_1)}\right) \cdot q_1 \cdot q_2}{4 \cdot l_1 \cdot l_2 \cdot \epsilon_0 \cdot \pi} | d=x \text{ and } l_1=1/2 \text{ and } l_2=3/2 \text{ and } q_1=1 \text{ and } q_2=2 \text{ and } \epsilon_0=1$$

$$\rightarrow \frac{-2 \cdot \ln\left(\frac{4 \cdot x \cdot (x+2)}{(2 \cdot x+1) \cdot (2 \cdot x+3)}\right)}{3 \cdot \pi}$$

$$f_1(x) := \frac{-2 \cdot \ln\left(\frac{4 \cdot x \cdot (x+2)}{(2 \cdot x+1) \cdot (2 \cdot x+3)}\right)}{3 \cdot \pi} \rightarrow \text{Done}$$



II.12 A thin electric rod is formed in the shape of an arc, which is a fragment given by the angle φ_0 of a circle of Radius R . The rod is uniformly charged with charge Q . Find the vector of the electric field strength for any point z_0 along the axis of the circle passing the centre and perpendicular to the plane of the circle. Discuss the result obtained ($\epsilon_0 = Q = R = 1$).

Solution:




Fig. II.12

According to Fig. II.12 the vector of the electric field strength can be calculated using the formula

$$\vec{E}(z, \varphi_0) = \frac{1}{4\pi\epsilon_0} \int_0^{\varphi_0} \frac{\lambda}{|\vec{r}|^3} \vec{r} R d\varphi,$$

where $\lambda = \frac{Q}{l_0}$ is the charge density with $l_0 = R\varphi_0$, and the vector $\vec{r} = (0 - x, 0 - y, z_0) = (-x, -y, z_0)$

with $x = R \cos(\varphi)$, $y = R \sin(\varphi)$. (\vec{r} is the direction vector from the wire to the point on the axis.)

We enter the relations given above and then calculate the field.

```
(%i1) l[0]:R*phi[0]$ lambda:Q/l[0]$ assume(R>0)$
      r_[:R*cos(phi),R*sin(phi),z[0]]$
(%o4) [-cos(phi)R, sin(phi)R, z_0]
(%i5) E_:lambda*R/(4*pi*epsilon[0])*
      integrate(trigsimp(1/sqrt(r_.r_)^3*r_),phi,0,phi[0])$
Is phi_0 positive, negative or zero?p;
(%o5) [ -frac(sin(phi_0)Q R, 4 pi epsilon_0 (R^2+z_0^2)^{3/2}),
        frac((1-cos(phi_0))Q R, 4 pi epsilon_0 (R^2+z_0^2)^{3/2}),
        frac(z_0 Q, 4 pi epsilon_0 (R^2+z_0^2)^{3/2}) ]
```

Now let us analyze the result obtained:

At the centre of the circle ($z_0 = 0$) we get the field:

```
(%i6) E_,z[0]=0;
(%o6) [ -frac(sin(phi_0)Q, 4 pi epsilon_0 R^2),
        frac((1-cos(phi_0))Q, 4 pi epsilon_0 R^2),
        0 ]
```

When the arc turns out to become the full circle ($\varphi_0 = 2\pi$) then only the z -component of field $E_$ is nonzero – as should be expected:

```
(%i7) E_,phi[0]=2*pi;
(%o7) [ 0, 0, frac(z_0 Q, 4 pi epsilon_0 (R^2+z_0^2)^{3/2}) ]
```

We extract the 3rd component of the field and we can easily see (from %o8) that at a large distance ($|z_0| \gg R$) the charged circle is equivalent to a point charge.

```
(%i8) E_z:E_[3];
(%o8) frac(z_0 Q, 4 pi epsilon_0 (R^2+z_0^2)^{3/2})
```

In infinity the field is zero.

```
(%i9) limit(E_z,z[0],inf);
(%o9) [ 0, 0, 0 ]
```

We find the extremal value of the field for the full ring:


```
(%i10) eq:diff(E_z,z[0])=0$
       solve(eq,z[0]);
(%o11) [ z_0 = -frac(R, sqrt(2)), z_0 = frac(R, sqrt(2)) ]
```

The field takes its extreme values in the points $\left(0, 0, \pm \frac{R\sqrt{2}}{2}\right)$.

We enter the given parameter values and then we can plot the graph of the z -component of the field.



```
(%i12) E_z,epsilon[0]=1,Q=1,R=1;
(%o12) 
$$\frac{z_0}{4\pi(z_0^2+1)^{3/2}}$$

(%i13) load(draw)$
(%i14) draw2d(title="Ez(z0)",explicit(%o12,z[0],-10,10));
```




Unfortunately Maxima does not provide such a wonderful tool as sliders for further investigation of the influence of parameters. Most other CAS do. We show the application of sliders to study the influence of charge Q and radius R as well.

Left hand side is a screen shot of DERIVE, right hand side a screen shot of TI-NspireCAS handheld. With TI-NspireCAS we can easily find the turning points, too.



Exercise: Consider two half circles of radii R mutually perpendicular with coinciding centres and uniformly charged with charge Q . Find the vector of the electric field strength and its magnitude in the circles' centre.

Hint: Use solution of problem II.12.



II.13 A thin ring of radius R has a uniformly distributed charge Q . Find the expression approximating the potential created by the ring in a point lying within the plane of the ring at a distance r from its centre, where $r \gg R$. (approximation of the field caused by a distant object).




Fig. II.13

Solution:

The potential caused by a charged element of the ring dl at a distance a from the ring is given by

$$d\Phi = \frac{1}{4\pi\epsilon_0} \frac{dQ}{a}, \text{ where } dQ = \lambda dl = \lambda R d\alpha \left(\lambda = \frac{Q}{2\pi R} \right).$$

Using the cosine rule we can express distance a as a function of angle α :

$$a = \sqrt{R^2 + r^2 - 2Rr \cos \alpha}.$$

The resultant potential is given by the following integral:

$$\Phi = \frac{1}{4\pi\epsilon_0} \int_0^{2\pi} \frac{\lambda R}{a} d\alpha = \frac{\lambda R}{4\pi\epsilon_0} \int_0^{2\pi} \frac{d\alpha}{\sqrt{R^2 + r^2 - 2Rr \cos \alpha}}.$$

Let's try to evaluate this integral.

```
(%i1) a:sqrt(R^2+r^2-2*R*r*cos(alpha))$ lambda:Q/(2*pi*R)$
(%i3) Phi:lambda*R/(4*pi*epsilon[0])*integrate(1/a,alpha,0,2*pi);
(%o3) \frac{Q}{8\pi^2\epsilon_0} \int_0^{2\pi} \frac{1}{\sqrt{R^2-2\cos(\alpha)rR+r^2}} d\alpha
```

We see that Maxima does not evaluate integral %o3 which means that we cannot expect a closed form of the integral, i.e. a combination of elementary functions. In order to approximate the potential we carry out the following steps:

- we introduce the parameter $\delta = \frac{R}{r}$ and expand the integrand as a Taylor series with respect to this parameter

```
(%i4) ex:ev(1/a,R=delta*r)$ taylor(ex,delta,0,2);
(%o5) T/ \frac{1}{r} + \frac{\cos(\alpha)\delta}{r} + \frac{(3\cos(\alpha)^2-1)\delta^2}{2r} + \dots
```

- and then we try integration again:

```
(%i6) Phiapprox:Q/(8*pi^2*epsilon[0])*integrate(%,alpha,0,2*pi);
(%o6) 
$$\frac{(\pi \delta^2 + 4 \pi) Q}{16 \epsilon_0 \pi^2 r}$$

```


So we get the final result of the above approximation (%o6). But let us resubstitute for δ and bring the result in a neat form (cancelling π needs the factor command!).

```
(%i7) ratsimp(ev(%,delta=R/r));
(%o7) 
$$\frac{\pi Q R^2 + 4 \pi r^2 Q}{16 \epsilon_0 \pi^2 r^3}$$

(%i8) factor(%);
(%o8) 
$$\frac{\pi Q (R^2 + 4 r^2)}{16 \epsilon_0 \pi^2 r^3}$$

```

We can work through the problem with TI-NspireCAS without any problems in the same way (including the cancellation ☺)!



$$\phi := \frac{\lambda \cdot r r}{4 \cdot \pi \cdot \epsilon_0} \int_0^{2 \cdot \pi} \frac{1}{a} d\alpha \rightarrow \frac{\int_0^{2 \cdot \pi} \frac{1}{\sqrt{-2 \cdot \cos(\alpha) \cdot r \cdot r + r^2 + r^2}} d\alpha \cdot q q}{8 \cdot \epsilon_0 \cdot \pi^2}$$

$$\text{ex} := \frac{1}{a} |_{r r = \delta \cdot r} \rightarrow \frac{1}{\sqrt{-2 \cdot \cos(\alpha) \cdot \delta + \delta^2 + 1} \cdot |r|}$$

$$\text{taylor}(\text{ex}, \delta, 2, 0) |_{r > 0} \rightarrow \frac{1}{r} + \frac{\delta \cdot \cos(\alpha)}{r} + \frac{\delta^2 \cdot (3 \cdot (\cos(\alpha))^2 - 1)}{2 \cdot r}$$

$$\frac{q q}{8 \cdot \epsilon_0 \cdot \pi^2} \int_0^{2 \cdot \pi} \left(\frac{1}{r} + \frac{\delta \cdot \cos(\alpha)}{r} + \frac{\delta^2 \cdot (3 \cdot (\cos(\alpha))^2 - 1)}{2 \cdot r} \right) d\alpha \rightarrow \frac{q q \cdot (\delta^2 + 4)}{16 \cdot \epsilon_0 \cdot \pi \cdot r}$$

$$\frac{q q \cdot (\delta^2 + 4)}{16 \cdot \epsilon_0 \cdot \pi \cdot r} |_{\delta = \frac{r r}{r}} \rightarrow \frac{q q \cdot (4 \cdot r^2 + r r^2)}{16 \cdot \epsilon_0 \cdot \pi \cdot r^3}$$

Exercise: Show that the electric field in the considered approximation is identical to the field caused by a point charge.

II.14 A circular plate of radius R is uniformly charged with a surface charge density σ .

- Calculate the field potential and electric field in a point lying a distance z away from the plate along the axis of symmetry perpendicular to the plate.
- By applying the appropriate *limit* calculate the function describing the field intensity resulting from an uniformly charged plate.
- Find an approximation for the field potential and intensity for the case that the radius of the plate is small with respect to the distance z .

Solution: The sketch of the situation is shown in Fig. II.14.




Fig. II.14

Similar to II.13 the potential resulting from charge Q in a point with distance l from the plate is given by

$$d\Phi = \frac{1}{4\pi\epsilon_0} \frac{dQ}{l}, \text{ where } l = \sqrt{r^2 + z^2} \text{ is the distance of charge } dQ \text{ from the observed point.}$$

$dQ = \sigma ds$ is the charge of the surface element of the plate. In polar coordinates $ds = r dr d\varphi$, therefore we have

$$dQ = \sigma ds = \sigma r dr d\varphi.$$

The potential caused by the plate is given by the integral

$$\Phi = \frac{\sigma}{4\pi\epsilon_0} \int_0^{2\pi} \int_0^R \frac{r}{l} dr d\varphi.$$

Taking into account the relation between field strength and field potential is – in the meanwhile well known – given by

$$\vec{E} = -\text{grad}(\Phi)$$

and the symmetry of the problem (only the z -component of the field strength can differ from zero), it is sufficient to calculate this component only:

$$\vec{E}_z = -\frac{d\Phi}{dz}.$$

a) We enter the expressions from above and calculate potential and field strength.

```
(%i1) l:sqrt(r^2+z^2)$ assume(R>0)$
(%i3) Phi:sigma/(4*pi*epsilon[0])*
      integrate(integrate(r/l,r,0,R),alpha,0,2*pi);
Is z positive or negative?p;
(%o3) 
$$\frac{\sigma(\sqrt{R^2+z^2}-z)}{2\epsilon_0}$$

(%i4) Ez:-diff(Phi,z);
(%o4) 
$$-\frac{\sigma\left(\frac{z}{\sqrt{R^2+z^2}}-1\right)}{2\epsilon_0}$$

```

b) To get the result for the plane ($R \rightarrow \infty$) it is sufficient to calculate the limit of Ez .

c)

```
(%i5) limit(Ez,R,inf);
(%o5) 
$$\frac{\sigma}{2\epsilon_0}$$

```

Remark: Of course, the electric field caused by a uniformly charged plane can be most easily obtained by applying *Gauss'* law.

Different methods of calculating the field strength deriving from the plain homogeneously loaded up still exist. We use the estimated electric field originating from the endless bar homogeneously loaded up (problem II.10 expression %o9).

The obtained result has the form $\frac{\lambda}{2\pi\epsilon_0 d}$, where λ

denotes the linear charge intensity and d the distance of the point of observation from the bar.

We try to use this result for our calculation. For this purpose we will divide the plane in stripes of widths dx .




Fig. II.14a

Charge dQ contained in surface element $ds = dx dy$ is $dQ = \sigma ds = \sigma dx dy$. On the other hand $dQ = \lambda dy$.

Comparing these two expressions for dQ we get $\lambda = \sigma dx$. Therefore in the solved exercise we have

$$d\vec{E} = \frac{1}{2\pi\epsilon_0} \frac{\sigma dx \vec{r}}{|\vec{r}|^2} \Rightarrow \vec{E} = \frac{\sigma}{2\pi\epsilon_0} \int_{-\infty}^{\infty} \frac{\vec{r}}{|\vec{r}|^2} dx \text{ with } \vec{r} = [-x, 0, z].$$

It is resulting from the symmetry of the problem, that only the z-component of the field should be nonzero. We will calculate the field and then read off this component:

```
(%i9) r_-: [-x, 0, z]
(%i16) E_: sigma / (2 * pi * epsilon[0]) * integrate(r_- / (r_- . r_-), x, minf, inf);
Is z zero or nonzero?p;
(%o16) [0, 0,  $\frac{\sigma}{2 \epsilon_0}$ ]
```

Let's perform the same calculation with DERIVE:

There we find another output for the third component of the field – and an undefined first component, too.

$$r_- := [-x, 0, z]$$

$$\frac{\sigma}{2 \cdot \pi \cdot \epsilon_0} \int_{-\infty}^{\infty} \frac{r_-}{|r_-|^2} dx = \left[?, 0, \frac{\sigma \cdot \text{SIGN}(z)}{2 \cdot \epsilon_0} \right]$$

In fact, calculation of the x-component is not trivial, namely in the calculation procedure

the integral $\int_{-\infty}^{\infty} \frac{x}{x^2 + z^2} dx$.

This integral can be transformed by substitution to the form $\int_{-\infty}^{\infty} \frac{du}{u}$. Not each CAS returns the satisfactory result. For example DERIVE gives "?" which is undefined. Maxima gives 0.

```
(%i11) integrate(1/u, u, minf, inf);
Principal Value
(%o11) 0
```

Below we show the calculation of the limit procedure stepwise:

$$\int_{-\infty}^{\infty} \frac{du}{u} = \lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow -\infty}} \int_a^{\epsilon} \frac{du}{u} + \lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow \infty}} \int_a^{\epsilon} \frac{du}{u} = \lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow \infty}} \ln \frac{\epsilon}{a} + \lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow \infty}} \ln \frac{a}{\epsilon} = -\lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow \infty}} \ln \frac{a}{\epsilon} + \lim_{\substack{\epsilon \rightarrow 0 \\ a \rightarrow \infty}} \ln \frac{a}{\epsilon} = 0.$$

For calculating the field strength from the plain homogeneously loaded up it is possible also to take advantage of the model describing the field strength from the dielectric circular ring homogeneously charged.

Make appropriate calculations. Use the result %08 of problem II.12.




Fig. II.14b

$$dE_z = \frac{z}{4\pi\epsilon_0} \frac{dQ}{(R^2 + z^2)^{3/2}} \text{ with } dQ = \sigma 2\pi R dR$$

$$E_z = \int_0^{\infty} dE_z = \int_0^{\infty} \frac{z}{4\pi\epsilon_0} \frac{\sigma 2\pi R dR}{(R^2 + z^2)^{3/2}} = \frac{z\sigma}{2\epsilon_0} \int_0^{\infty} \frac{R dR}{(R^2 + z^2)^{3/2}}$$

```
(%i1) assume(z>0)$
(%i2) Ez: (z*sigma/(2*epsilon[0]))*'integrate(R/(R^2+z^2)^(3/2),R,0,inf)=
(z*sigma/(2*epsilon[0]))*'integrate(R/(R^2+z^2)^(3/2),R,0,inf);
(%o2) 
$$\frac{\sigma z \int_0^{\infty} \frac{R}{(R^2+z^2)^{3/2}} dR}{2 \epsilon_0} = \frac{\sigma}{2 \epsilon_0}$$

```

- d) Approximations for potential and electric field (for $R \ll z$) can be obtained by expanding the respective expressions as Taylor series with respect to the parameter $\delta = \frac{R}{z}$.

```
(%i6) Pi_approx:taylor(ev(Phi,R=delta*z),delta,0,2);
(%o6) /T/  $\frac{\sigma z \delta^2}{4 \epsilon_0} + \dots$ 
(%i7) E_approx:taylor(ev(Ez,R=delta*z),delta,0,2);
(%o7) /T/  $\frac{\sigma \delta^2}{4 \epsilon_0} + \dots$ 
```

Remark: In a similar way we can solve the analogous problem for a ring (a circle with the centre removed) with inner radius R_1 and outer radius R_2 .

II.15 A fragment of a thin dielectric circular arc characterized by two radii R_1 and R_2 , and by the angle φ_0 (see Fig.II.15) is charged with a radial dependent surface density $\sigma(R) = \frac{k}{R}$. Find the components of the vector of the electric strength. Analyze the obtained results.

Solution:

The present problem is more general than the previous one. Because of the geometry of the problem it is convenient to use polar coordinates. The strength of the electric field can be expressed by the integral

$$\vec{E}(z, R_1, R_2, \varphi_0) = \frac{1}{4\pi\epsilon_0} \int_0^{\varphi_0} \int_{R_1}^{R_2} \frac{\sigma(r)\vec{r}}{|\vec{r}|^3} ds = \frac{1}{4\pi\epsilon_0} \int_0^{\varphi_0} \int_{R_1}^{R_2} \frac{\sigma(r)\vec{r}}{|\vec{r}|^3} R dR d\varphi$$

where (see Fig. II.15) vector \vec{r} has components $(0-x, 0-y, z-0) = (-x, -y, z)$.

The relationship between Cartesian and polar coordinates is

$$x = R \cos(\varphi), \quad y = R \sin(\varphi).$$




Fig. II.15

We enter the expressions given above

```
(%i1) sigma(R):=k/R$
(%i2) x:R*cos(phi)$ y:R*sin(phi)$ r_:[-x,-y,z]$
```

and then try to calculate the field strength followed by extracting the desired components:

```
(%i5) E_:1/(4*pi*epsilon[0])*
      integrate(
        integrate(sigma(R)/sqrt(r_.r_)^3*r_*R,
          R,R1,R2),phi,0,Phi[0])$
Is R2-R1 positive, negative or zero?p;
Is phi_0 positive, negative or zero?p;
(%i6) Ex:E_[1];
(%o6) 
$$\frac{k \left( \sin(\phi_0) \sqrt{R2^2 + z^2} - \sin(\phi_0) \sqrt{R1^2 + z^2} \right)}{4 \pi \epsilon_0 \sqrt{R1^2 + z^2} \sqrt{R2^2 + z^2}}$$

(%i7) Ey:E_[2];
(%o7) -

$$\frac{k \left( \left( (\cos(\phi_0) - 1) R1^2 + (\cos(\phi_0) - 1) z^2 \right) \sqrt{R2^2 + z^2} + \sqrt{R1^2 + z^2} \left( (1 - \cos(\phi_0)) R2^2 + (1 - \cos(\phi_0)) z^2 \right) \right)}{4 \pi \epsilon_0 \left( (R1^2 + z^2) R2^2 + z^2 R1^2 + z^4 \right)}$$

(%i8) Ez:E_[3];
(%o8) 
$$\frac{k z \left( \left( \phi_0 R1^2 + \phi_0 z^2 \right) R2 \sqrt{R2^2 + z^2} + \sqrt{R1^2 + z^2} \left( -\phi_0 R1 R2^2 - \phi_0 z^2 R1 \right) \right)}{4 \pi \epsilon_0 \left( \left( z^2 R1^2 + z^4 \right) R2^2 + z^4 R1^2 + z^6 \right)}$$

```

As these are very bulky expressions we can hope to obtain more comfortable expressions by first simplifying the integrand applying `trigsimp`.

And indeed, it works as hoped (and expected?).

```
(%i9) E_:1/(4*%pi*epsilon[0])*
      integrate(
          integrate(
              trigsimp(sigma(R)/sqrt(r_.r_)^3*r_*R),
              R,R1,R2),phi,0,Phi[0])$
Is R2-R1 positive, negative or zero?p;
Is phi_0 positive, negative or zero?p;
(%i10) Ex_:E_[1];
      sin(phi_0) k \left( \frac{1}{\sqrt{R1^2+z^2}} - \frac{1}{\sqrt{R2^2+z^2}} \right)
(%o10) -----
      4 pi epsilon_0
(%i11) Ey_:E_[2];
      (1-cos(phi_0)) k \left( \frac{1}{\sqrt{R1^2+z^2}} - \frac{1}{\sqrt{R2^2+z^2}} \right)
(%o11) -----
      4 pi epsilon_0
(%i12) Ez_:E_[3];
      phi_0 k z \left( \frac{R2\sqrt{R2^2+z^2}}{z^2 R2^2+z^4} - \frac{R1\sqrt{R1^2+z^2}}{z^2 R1^2+z^4} \right)
(%o12) -----
      4 pi epsilon_0
```

In case of a full circle ($\phi_0 = 2\pi$) we get:

```
(%i13) E_,Phi[0]=2*%pi;
      k z \left( \frac{R2\sqrt{R2^2+z^2}}{z^2 R2^2+z^4} - \frac{R1\sqrt{R1^2+z^2}}{z^2 R1^2+z^4} \right)
(%o13) [0, 0, -----]
      2 epsilon_0
```

As one would expect only the z-component of the field is results as nonzero.

In case of an infinite plane ($R_1 = 0, R_2 = \infty$) we obtain:

```
(%i14) limit(ev(E_,R1=0,Phi[0]=2*%pi),R2,inf);
(%o14) [0, 0, \frac{k}{2 epsilon_0 z}]
```

So we received the well known result that the field strength is inverse proportional to the distance from the plane (see also problem II.10).

What will happen in case of a constant surface density?

```
(%i1) x:R*cos(phi)$ y:R*sin(phi)$ r_:[-x,-y,z]$
(%i5) E_:1/(4*%pi*epsilon[0])*
      integrate(
          integrate(trigsimp(sigma/sqrt(r_.r_)^3*r_*R),
              R,R1,R2),phi,0,Phi[0])$
Is R2-R1 positive, negative or zero?p;
Is phi_0 positive, negative or zero?p;
(%i8) assume(z>0);
(%o8) [z>0]
```

```
(%i12) ratsimp(ev(E_,Phi[0]=2*%pi,R1=0));
(%o12) [0,0, $\frac{\sigma\sqrt{R2^2+z^2}-\sigma z}{2\varepsilon_0\sqrt{R2^2+z^2}}$ ]
(%i11) limit(%o10,R2,inf);
(%o11) [0,0, $\frac{\sigma}{2\varepsilon_0}$ ]
```

As should be expected for the full circle of radius R_2 ($R_1 = 0$) we obtain %o12, which is the same as that obtained in the previous problem (II.14, expression %o4).

For the infinite plane we get the well known result (%o11) that field strength does not depend on the distance of the observed point from the plane.

II.16 A non-conductive rectangular plate of dimension $a \times b$ lying in the xy -plane with its centre in the origin is charged. The surface density depends on the position vector in the following way: $\sigma(y) = \sigma_0 \left(1 - \frac{y}{b}\right)$. Find the vector of the electric field strength caused by this plate in a generic point $P(0,0,z_0)$.

Solution:

Electric field strength in point P created by an element $dx dy$ of the plate can be expressed according *Coulomb's Law* in the form

$$d\vec{E} = \frac{1}{4\pi\varepsilon_0} \frac{\sigma(\vec{r})}{(\vec{r}_0 - \vec{r})^2} \frac{(\vec{r}_0 - \vec{r})}{|\vec{r}_0 - \vec{r}|} ds = \frac{1}{4\pi\varepsilon_0} \frac{\sigma(\vec{r})}{(\vec{r}_0 - \vec{r})^2} \frac{(\vec{r}_0 - \vec{r})}{|\vec{r}_0 - \vec{r}|} dx dy,$$

where $\vec{r}(x, y, z)$, $\vec{r}_0(0, 0, z_0)$ denote the position vectors of the element of the plate $dx dy$ and the observed point, respectively.




Fig. II.16

The resultant vector of the electric field strength is obtained by integrating the above expression over the surface of the plate

$$\vec{E} = \frac{1}{4\pi\varepsilon_0} \int_{-a/2}^{a/2} \int_{-b/2}^{b/2} \frac{\sigma(\vec{r})}{(\vec{r}_0 - \vec{r})^2} \frac{(\vec{r}_0 - \vec{r})}{|\vec{r}_0 - \vec{r}|} dy dx = \frac{1}{4\pi\varepsilon_0} \int_{-a/2}^{a/2} \int_{-b/2}^{b/2} \sigma(\vec{r}) \frac{(\vec{r}_0 - \vec{r})}{|\vec{r}_0 - \vec{r}|^3} dy dx.$$

The above integral seems to be too hard for wxMaxima, therefore we apply DERIVE.

We enter the data and the define the components:

$$\#1: \left[\sigma := \sigma_0 \cdot \left(1 - \frac{y}{b} \right), r_- := [x, y, 0], r_{0-} := [0, 0, z_0] \right]$$

$$\#2: E_- := \frac{1}{4 \cdot \pi \cdot \epsilon_0} \cdot \int_{-a/2}^{a/2} \int_{-b/2}^{b/2} \frac{\sigma}{|r_{0-} - r_-|^3} \cdot (r_{0-} - r_-) dy dx$$

$$\#3: \left[E_x := E_{-1}, E_y := E_{-2}, E_z := E_{-3} \right]$$

The components are:

$$\#4: E_x = 0$$


$$\#5: E_y = \frac{a \cdot \sigma_0 \cdot \text{LN} \left(\frac{(\sqrt{(a^2 + b^2 + 4 \cdot z_0^2)} + b)^2}{a^2 + 4 \cdot z_0^2} \right)}{4 \cdot \pi \cdot b \cdot \epsilon_0} - \frac{z_0 \cdot \sigma_0 \cdot \text{ATAN} \left(\frac{a \cdot b}{2 \cdot z_0 \cdot \sqrt{(a^2 + b^2 + 4 \cdot z_0^2)}} \right)}{\pi \cdot b \cdot \epsilon_0}$$

$$\#6: E_z = \frac{\sigma_0 \cdot \text{ATAN} \left(\frac{a \cdot b}{2 \cdot z_0 \cdot \sqrt{(a^2 + b^2 + 4 \cdot z_0^2)}} \right)}{\pi \cdot \epsilon_0}$$

We can plot the components depending on z_0 for special data:

$$\#7: [\epsilon_0 := 1, \sigma_0 := 100, a := 1, b := 2]$$

$$\#8: E_-$$



II.17 An electric charge Q is distributed uniformly on a semi-sphere of radius R . Find the vector of the electric field in an arbitrary point on the axis of symmetry of the sphere.

Find the vector of the electric field in the centre of the sphere and in point $B(0,0,-R)$.

Solution:

The contribution to the vector of the electric field \overline{dE} produced by an element dS of the semi-sphere is given by

$$\overline{dE} = \frac{1}{4\pi\epsilon_0} \frac{dQ}{|\vec{r}|^3} \vec{r} = \frac{1}{4\pi\epsilon_0} \frac{\sigma dS}{|\vec{r}|^3} \vec{r},$$

where σ denotes the surface charge density, $\sigma = \frac{Q}{2\pi R^2}$,

and at the observed point $(0,0,z_0)$ the vector \vec{r} is

$$\vec{r} = \vec{r}(0-x, 0-y, z_0-z) = \vec{r}(-x, -y, z_0-z).$$




Fig. II.17

The resultant field vector is obtained by integrating the above expression over the whole semi-sphere

$$\vec{E} = \frac{1}{4\pi\epsilon_0} \iint_S \frac{\sigma}{|\vec{r}|^3} \vec{r} dS = \frac{\sigma}{4\pi\epsilon_0} \iint_S \frac{\vec{r}}{|\vec{r}|^3} dS.$$

If we use spherical coordinates

$$x = R \sin \delta \cos \varphi, \quad y = R \sin \delta \sin \varphi, \quad z = R \cos \delta,$$

then the surface element dS is given by:

$$dS = R^2 \sin \delta d\delta d\varphi$$

and the field vector in a point $(0,0,z_0)$ on the z axis can be written in the form:

$$\vec{E}(z_0) = \frac{\sigma}{4\pi\epsilon_0} \int_0^{\pi/2} \int_0^{2\pi} \frac{\vec{r}}{|\vec{r}|^3} R^2 \sin \delta d\varphi d\delta = \frac{\sigma R^2}{4\pi\epsilon_0} \int_0^{\pi/2} \int_0^{2\pi} \frac{\vec{r}}{|\vec{r}|^3} \sin \delta d\varphi d\delta.$$

We enter the expression above and try to calculate the field vector (%06):

```

(%i1) sigma:Q/(2*pi*R^2)$
(%i2) x:R*sin(theta)*cos(phi)$ y:R*sin(theta)*sin(phi)$
z:R*cos(theta)$
(%i5) r_-:[-x,-y,z0]$
(%i6) E_:sigma*R^2/(4*pi*eps0)*
integrate(
integrate(r_/sqrt(r_._r_)^3*sin(theta),
phi,0,2*pi),
theta,0,%pi/2);
Is z0 positive or negative?p;
(%o6) [0,0,frac{z0 Q}{4 pi eps0 (z0 R^2+z0^3)}]
(%i7) E_:factor(E_);
(%o7) [0,0,frac{Q}{4 pi eps0 (R^2+z0^2)}]

```

factor(E_) gives the simplest form of the result (cancelling z₀):

To get the field in point B and in the centre we have to perform the respective substitutions (z₀ = -R and z₀ = 0):

```

(%i8) ev(E_,z0=-R);
(%o8) [0,0,frac{Q}{8 pi eps0 R^2}]
(%i9) ev(E_,z0=0);
(%o9) [0,0,frac{Q}{4 pi eps0 R^2}]

```

Note: If we would like to calculate the field in the centre using result %o6 we need to apply the limit command (see the error message as answer in %i11):

```

(%i10) limit(%o6,z0,0);
(%o10) [0,0,frac{Q}{4 pi eps0 R^2}]
(%i11) ev(%o6,z0=0);
expt: undefined: 0 to a negative exponent.
-- an error. To debug this try: debugmode(true)

```

TI-NspireCAS has no problems performing the double integration:

$\sigma := \frac{q}{2 \cdot \pi \cdot r^2}$
 $x := r \cdot \sin(\delta) \cdot \cos(\varphi)$
 $y := r \cdot \sin(\delta) \cdot \sin(\varphi)$
 $z := r \cdot \cos(\delta)$
 $r_- := [-x \ -y \ z0]$

$[-\sin(\delta) \cdot \cos(\varphi) \cdot r \ -\sin(\delta) \cdot \sin(\varphi) \cdot r \ z0]$

$$e_- := \frac{\sigma \cdot r^2}{4 \cdot \pi \cdot \epsilon_0} \int_0^{\frac{\pi}{2}} \int_0^{2 \cdot \pi} \left(\frac{r_-}{(\text{norm}(r_-))^3} \cdot \sin(\delta) \right) d\varphi d\delta |z_0 > 0$$

$$\begin{bmatrix} 0 & 0 & \frac{q}{4 \cdot \epsilon_0 \cdot \pi \cdot (r^2 + z_0^2)} \end{bmatrix}$$

II.18 A cylindrical dielectric layer characterized by two radii R_1 , R_2 and height H is uniformly charged with a charge Q . Find the vector of the electric field

- on the axis of symmetry,
- in the centre of the base circle (see Fig. II.18).




Fig. II.18

Solution:

The contribution of an element dV of the hollow cylinder to the field vector $d\vec{E}$ in point \vec{r} , is given by

$$d\vec{E} = \frac{1}{4\pi\epsilon_0} \frac{dQ}{|\vec{r}'|^3} \vec{r}' = \frac{1}{4\pi\epsilon_0} \frac{\rho}{|\vec{r}'|^3} \vec{r}' dV,$$

where ρ denotes the charge density $\rho = \frac{Q}{\pi(R_2^2 - R_1^2)H}$.

The resultant field vector is the result of integrating $d\vec{E}$ over the whole cylinder

$$\vec{E} = \frac{1}{4\pi\epsilon_0} \iiint_V \frac{\rho}{|\vec{r}'|^3} \vec{r}' dV = \frac{\rho}{4\pi\epsilon_0} \iiint_V \frac{\vec{r}'}{|\vec{r}'|^3} dV.$$

Due to the symmetry of the system, it is convenient to use cylindrical coordinates: According to Fig II.18 we can write

$$\vec{r}' = \vec{r}'(0 - x, 0 - y, z_0 - z) = \vec{r}'(-x, -y, z_0 - z) \text{ with}$$

$$x = R \cos \varphi, \quad y = R \sin \varphi, \quad dV = R dR d\varphi dz.$$

The resultant vector of the electric field is of the form

$$\vec{E} = \frac{\rho}{4\pi\epsilon_0} \int_0^H \int_0^{2\pi} \int_{R_1}^{R_2} \frac{\vec{r}'}{|\vec{r}'|^3} R dR d\varphi dz.$$

We enter the expressions given above:

```
(%i1) rho:Q/(%pi*(R2^2-R1^2)*H) $ assume(R1>0, R2>R1,H>0) $
(%i3) x:R*cos(phi) $ y:R*sin(phi) $ r_:[-x,-y,z0-z] $
```

- a) At first we try to calculate the strength of the field using the `trigsimp` command within the integration.


```
(%i6) fv:r_/sqrt(r_.r_)^3*R$
(%i7) E_:rho/(4*%pi*eps0)*
      integrate(
        integrate(
          integrate(trigsimp(fv),R,R1,R2),
            phi,0,2*%pi),
          z,0,H);
Is z0 zero or nonzero?p;
(%o7) [0,0,

$$\frac{Q(\sqrt{R2^2+H^2-2z0H+z0^2}-\sqrt{R2^2+z0^2}-\sqrt{R1^2+H^2-2z0H+z0^2}+\sqrt{R1^2+z0^2})}{2\pi\epsilon_0 H(R2^2-R1^2)}$$
]
```

Graphic representation is obtained as follows.

```
(%i8) plE_:ev(E_,eps0=1,Q=1,R1=1,R2=2,H=5);
(%o8) [0,0,

$$\frac{\sqrt{z0^2-10z0+29}-\sqrt{z0^2-10z0+26}-\sqrt{z0^2+4}+\sqrt{z0^2+1}}{30\pi}$$
];
(%i9) plot2d(plE_,[z0,-10,10]);
(%o9) [C:/Users/Josef/maxout.gnuplot]
```



- b) In the point of interest ($z_0 = 0$) the field takes a simpler form:

```
(%i10) ev(E_,z0=0);
(%o10) [0,0,

$$\frac{Q(\sqrt{R2^2+H^2}-R2-\sqrt{R1^2+H^2}+R1)}{2\pi\epsilon_0 H(R2^2-R1^2)}$$
]
```

The many parameters invite introducing sliders for analyzing their influence on the strength of the field. We do this supported by TI-NspireCAS (we could also use DERIVE for this purpose). We can also ask for the position of the maximum strength.

$$\rho := \frac{q}{\pi \cdot (r1^2 - r2^2) \cdot h}$$

$$r_- := [-r \cdot \cos(\varphi) \quad -r \cdot \sin(\varphi) \quad z0 - z]$$

$$\hat{r}_v := \frac{r_-}{(\text{norm}(r_-))^3}$$


$$e_- := \frac{\rho}{4 \cdot \pi \cdot \epsilon0} \int_0^h \int_0^{2 \cdot \pi} \int_{r1}^{r2} \hat{r}_v \, dr \, d\varphi \, dz | z0 > 0$$

$$e_- [1,3] | z0 = x \text{ and } q = 1 \text{ and } \epsilon0 = 1 \text{ and } h = 5 \text{ and } r1 = 1 \text{ and } r2 = 2$$

$$f1(x) := \frac{\sqrt{x^2 - 10 \cdot x + 29} - \sqrt{x^2 - 10 \cdot x + 26} - \sqrt{x^2 + 4} + \sqrt{x^2 + 1}}{30 \cdot \pi}$$

$$f2(x) := e_- [1,3] | z0 = x \text{ and } q = 1 \text{ and } \epsilon0 = 1$$

Calculation is easy done. The screen below shows the graph from above (black) and the graph variable by moving the sliders (red). The maximum strength of the black graph is given for $z_0 \approx 5.1$ using the TI's Analyze-tool. It cannot be calculated exactly.



II.19 Calculate potential and electric field caused by a uniformly charged sphere of charge Q and radius R . Try to apply various ways of solutions. Present the results in form of a graph.

Solution:

Method One

The electric potential $d\Phi$ caused by a point charge dQ at a distance l from this charge (see Fig. II.19) is equal to

$$d\Phi = \frac{1}{4\pi\epsilon_0} \frac{dQ}{|\vec{r} - \vec{r}'|} = \frac{1}{4\pi\epsilon_0} \frac{\rho dV}{|\vec{r} - \vec{r}'|},$$

where $\vec{r} = \vec{r}(x, y, z)$, $\vec{r}' = \vec{r}'(x', y', z')$. $\rho = \frac{Q}{\frac{4}{3}R^3\pi}$ denotes the volume charge density and dV denotes

the volume element.




Fig. II.19

The resultant potential is given by the integral

$$\Phi = \int_V d\Phi = \frac{\rho}{4\pi\epsilon_0} \int_V \frac{dV}{|\vec{r} - \vec{r}'|}.$$

In this problem it is useful to change from Cartesian coordinates to spherical ones.

$$x = r \sin \theta \cos \varphi, \quad y = r \sin \theta \sin \varphi, \quad z = r \cos \theta.$$

The volume element dV in spherical coordinates is given by

$$dV = r^2 \sin \theta dr d\theta d\varphi.$$

The potential integral finally gets the form

$$\Phi = \frac{\rho}{4\pi\epsilon_0} \int_0^{2\pi} \int_0^{\pi} \int_0^R \frac{r^2 \sin \theta}{|\vec{r} - \vec{r}'|} dr d\theta d\varphi.$$

When we try calculating the above triple integral directly we will meet some difficulties. However, we can make use of the symmetry of the system and locate the observed point on the z -axis:

$$\vec{r}' = \vec{r}_p = \vec{r}_p(0, 0, zp).$$

Note: variable zp plays the role of the distance l of the observed point form the centre of the sphere.

We start the evaluation procedure:

```
(%i1) rho:Q/(4/3*pi*R^3)$ assume(R>0)$
(%i3) x:r*sin(theta)*cos(phi)$ y:r*sin(theta)*sin(phi)$ z:r*cos(theta)$
(%i6) r_: [x,y,z]$ rp_: [0,0,zp]$ assume(zp>0)$
(%i9) p:trigsimp(r^2*sin(theta)/sqrt((rp_-r_).(rp_-r_)))$
(%i10) Phi:rho/(4*pi*eps0)*
        integrate(
            integrate(
                integrate(p,theta,0,%pi),
                phi,0,2*pi),
            r,0,R)$
```

Maxima encountered a Lisp error:

```
Control stack exhausted (no more space for function call frames).
This is probably due to heavily nested or infinitely recursive function
calls, or a tail call that SBCL cannot or has not optimized away.
PROCEED WITH CAUTION.
Automatically continuing.
To enable the Lisp debugger set *debugger-hook* to nil.
```

We notice that Maxima is unable to perform the integration. Let's try to assist the program by stepwise integrating starting with the innermost integral:

```
(%i11) Phi:integrate(p,theta,0,%pi);
(%o11) r^2 * ( (sqrt(zp^2+2*r*zp+r^2) * sqrt(zp^2-2*r*zp+r^2)) / (r*zp) )
(%i12) Phi:integrate(Phi,phi,0,2*pi);
(%o12) 2*pi*r^2 * ( (sqrt(zp^2+2*r*zp+r^2) * sqrt(zp^2-2*r*zp+r^2)) / (r*zp) )
(%i13) Phi:integrate(Phi,r,0,R);
(%o13) pi * ( (-2*R^2+zp*R+zp^2) * sqrt(R^2-2*zp*R+zp^2) + sqrt(R^2+2*zp*R+zp^2) * (2*R^2+zp*R-zp^2) ) / (3*zp)
(%i14) Phi:rho/(4*pi*eps0)*Phi;
(%o14) rho * ( (-2*R^2+zp*R+zp^2) * sqrt(R^2-2*zp*R+zp^2) + sqrt(R^2+2*zp*R+zp^2) * (2*R^2+zp*R-zp^2) ) / (16*pi*eps0*zp*R^3)
```

Now we get a result for Φ which can be further simplified:

```
(%i15) Phi:ratsimp(scanmap(factor,Phi));
(%o15) (2*Q*R^2-zp*Q*R-zp^2*Q) * |R-zp| * (-2*Q*R^3-3*zp*Q*R^2+zp^3*Q) / (16*pi*eps0*zp*R^3)
```

The absolute value expression indicates that we have to distinguish between two cases:

$R \geq zp$ (observed point is inside the sphere) and $R < zp$ (observed point is outside the sphere).

```
(%i16) assume(zp<R)$ Phi_in:ratsimp(Phi);
(%o17) 
$$\frac{3 Q R^2 - z p^2 Q}{8 \pi \text{eps}0 R^3}$$

(%i18) forget(zp<R)$ assume(zp>R)$ Phi_out:ratsimp(Phi);
(%o20) 
$$\frac{Q}{4 \pi \text{eps}0 z p}$$

```

Thus we can enter the potential in form of a piecewise defined function $\Phi(zp)$:

```
(%i21) Phi(zp):=if zp<=R then (3*Q*R^2-zp^2*Q)/(8*pi*eps0*R^3)
      else Q/(4*pi*eps0*zp);
(%o21) 
$$\Phi(zp):=\text{if } zp \leq R \text{ then } \frac{3 Q R^2 - z p^2 Q}{8 \pi \text{eps}0 R^3} \text{ else } \frac{Q}{4 \pi \text{eps}0 z p}$$

```

Field strength calculation uses the well known relation $\vec{E} = -\text{grad}(\Phi)$. In our problem we need only consider differentiation wrt zp .

```
(%i22) Ez_in:-diff(Phi_in,zp);
(%o22) 
$$\frac{z p Q}{4 \pi \text{eps}0 R^3}$$

(%i23) Ez_out:-diff(Phi_out,zp);
(%o23) 
$$\frac{Q}{4 \pi \text{eps}0 z p^2}$$

(%i24) E_(zp):=[0,0,if zp<=R then (zp*Q)/(4*pi*eps0*R^3)
      else Q/(4*pi*eps0*zp^2) ];
(%o24) 
$$E_(zp):=[0,0,\text{if } zp \leq R \text{ then } \frac{z p Q}{4 \pi \text{eps}0 R^3} \text{ else } \frac{Q}{4 \pi \text{eps}0 z p^2}]$$

```

Inside the sphere field strength increases proportional to the distance from the centre of the sphere ($\propto zp$), whereas outside it decreases inversely proportional to this distance ($\propto \frac{1}{z p^2}$).

It is no problem to perform the integration with DERIVE:

$$[R : \in \text{Real} (0, \infty), zp : \in \text{Real} (0, \infty)]$$

$$\Phi := \frac{1}{4 \cdot \pi \cdot \text{eps}0} \cdot \int_0^\pi \int_0^{2 \cdot \pi} \int_0^R \frac{\rho \cdot r^2 \cdot \text{SIN}(\theta)}{|r p_- - r_-|} dr d\phi d\theta$$

$$\Phi := \frac{Q \cdot (2 \cdot R^3 + 3 \cdot R^2 \cdot z p - z p^3)}{16 \cdot \pi \cdot R^3 \cdot z p \cdot \text{eps}0} - \frac{Q \cdot (2 \cdot R^2 - R \cdot z p - z p^2) \cdot |R - z p|}{16 \cdot \pi \cdot R^3 \cdot z p \cdot \text{eps}0}$$

But I came across problems working with TI-NspireCAS. I could perform the integration, but I could not obtain the "inside" result.

| | | |
|---|---|---|
| ⚠ | $\frac{-3 \cdot q \cdot \int_0^{rr} (r \cdot (r - zp - (r + zp))) dr}{8 \cdot rr^3 \cdot zp \cdot \epsilon 0 \cdot \pi}$ | $\frac{3 \cdot q}{8 \cdot rr \cdot \epsilon 0 \cdot \pi}$ |
| ⚠ | $\frac{-3 \cdot q \cdot \int_0^{rr} (r \cdot (-r + zp - (r + zp))) dr}{8 \cdot rr^3 \cdot zp \cdot \epsilon 0 \cdot \pi}$ | $\frac{q}{4 \cdot zp \cdot \epsilon 0 \cdot \pi}$ |

I wrote to Michel Beaudin (Canada) and he knew how to do overcome the problem. See Michel's answer and advice as well:

Josef, note that if the point is inside the sphere, we have $R > zp$. So in order to compute the value of « phi » in your Nspire file, you need to split the integral into 2 parts because you don't know if r is between 0 and zp or between zp and rr . Doing this, Nspire gets the same results as good old Derive (see the file). In Derive, we don't need to do this because Derive is able to compute the whole integral because Derive is using a special rule of integration involving signum functions :
 $\text{int}(f(x) \cdot \text{sign}(a \cdot x + b)) = \dots\dots$

| | | |
|---|---|--|
| ⚠ | $\rho := \frac{q}{\frac{4}{3} \cdot \pi \cdot rr^3}$ | $\frac{3 \cdot q}{4 \cdot rr^3 \cdot \pi}$ |
| | $x := r \cdot \sin(\theta) \cdot \cos(\phi); y := r \cdot \sin(\theta) \cdot \sin(\phi); z := r \cdot \cos(\theta)$ | $\cos(\theta) \cdot r$ |
| | $r_ := [x \ y \ z]; rp_ := [0 \ 0 \ zp]$ | $[0 \ 0 \ zp]$ |
| | $\text{phi} := \frac{\rho}{4 \cdot \pi \cdot \epsilon 0} \int_0^{rr} \int_0^{2 \cdot \pi} \int_0^{\pi} \frac{r^2 \cdot \sin(\theta)}{\text{norm}(rp_ - r_)} d\theta d\phi dr r > 0 \text{ and } zp > 0$ | $\frac{-3 \cdot q \cdot \int_0^{rr} (r \cdot (r - zp - r - zp)) dr}{8 \cdot rr^3 \cdot zp \cdot \epsilon 0 \cdot \pi}$ |
| | © outside sphere: $zp > R$. | |
| ⚠ | $\frac{-3 \cdot q \cdot \int_0^{rr} (r \cdot (-r + zp - r - zp)) dr}{8 \cdot rr^3 \cdot zp \cdot \epsilon 0 \cdot \pi}$ | $\frac{q}{4 \cdot zp \cdot \epsilon 0 \cdot \pi}$ |
| | © inside sphere: split the integral in $0 \leq r < zp$ and $zp \leq r \leq rr$ | |
| ⚠ | $\frac{-3 \cdot q \cdot \int_0^{zp} (r \cdot (zp - r - (r + zp))) dr + 3 \cdot q \cdot \int_{zp}^{rr} (r \cdot (r - zp - (r + zp))) dr}{8 \cdot rr^3 \cdot zp \cdot \epsilon 0 \cdot \pi}$ | $\frac{q \cdot (3 \cdot rr^2 - zp^2)}{8 \cdot rr^3 \cdot \epsilon 0 \cdot \pi}$ |

Many thanks to Michel.

The plots are following on page 71.

Method Two

We can solve the problem by integrating *Poisson's* equation (see page 14). This equation is of form

$$\nabla^2\Phi = -\frac{\rho}{\epsilon_0} \text{ inside the sphere and}$$

$$\nabla^2\Phi = 0 \text{ outside the sphere,}$$

where ∇^2 denotes the *Laplace* operator. When using spherical coordinates due to radial symmetry such an operator simplifies to

$$\frac{1}{r^2} \left(\frac{d}{dr} \left(r^2 \frac{d\Phi}{dr} \right) \right).$$

We enter the equation for the "inside" potential and proceed (comments in text cells):

```
(%i2) eq1:1/r^2*'diff(r^2*diff(phi(r),r),r)=-rho/eps0;
```

$$(\%o2) \frac{\frac{d}{dr} \left(r^2 \left(\frac{d}{dr} \varphi(r) \right) \right)}{r^2} = -\frac{\rho}{\epsilon_0}$$

after multiplying by r^2 integrations is an easy job:

```
(%i4) integrate(eq1*r^2,r);
```

$$(\%o4) r^2 \left(\frac{d}{dr} \varphi(r) \right) = \%c1 - \frac{r^3 \rho}{3 \epsilon_0}$$

we divide by r^2 and integrate again:

```
(%i5) integrate(%/r^2,r);
```

$$(\%o5) \varphi(r) = -\frac{r^2 \rho}{6 \epsilon_0} - \frac{\%c1}{r} + \%c2$$

The potential cannot tend to infinity as $r \rightarrow 0$, thus the constant $\%c1$ must equal zero. Thus the potential inside the sphere is

```
(%i8) phi_in(r):=- (r^2*rho)/(6*eps0)+%c2;
```

$$(\%o8) \text{phi_in}(r) := -\frac{r^2 \rho}{6 \epsilon_0} + \%c2$$

Now let's work out the "outside" potential:

```
(%i5) eq2:1/r^2*'diff(r^2*diff(phi(r),r),r)=0;
```

$$(\%o5) \frac{\frac{d}{dr} \left(r^2 \left(\frac{d}{dr} \varphi(r) \right) \right)}{r^2} = 0$$

analogous to the previous calculation:

```
(%i6) ratsimp(integrate(eq2*r^2,r));
```

$$(\%o6) r^2 \left(\frac{d}{dr} \varphi(r) \right) = \%c3$$

```
(%i7) integrate(%/r^2,r);
```

$$(\%o7) \varphi(r) = \%c4 - \frac{\%c3}{r}$$

```
(%i8) phi_out(r):=%c4-%c3/r;
(%o8) phi_out(r):=%c4- $\frac{\%c3}{r}$ 
```

It remains to determine the integration constants %c2 to %c4, from the boundary conditions:

- the potential must tend to 0 for $r \rightarrow \infty$ and
- potential and strength must be continuous on the surface of the sphere

which leads to the following three equations which can easily be solved:

```
(%i9) eq3:limit(phi_out(r),r,inf)=0;
(%o9) %c4=0
(%i10) eq4:limit(phi_in(r),r,R)=limit(phi_out(r),r,R);
(%o10)  $\frac{\rho R^2 - 6 \%c2 \text{eps}0}{6 \text{eps}0} = \frac{\%c4 R - \%c3}{R}$ 
(%i11) eq5:limit(diff(phi_in(r),r),r,R)=limit(diff(phi_out(r),r),r,R);
(%o11)  $\frac{\rho R}{3 \text{eps}0} = \frac{\%c3}{R^2}$ 
(%i12) solve([eq3,eq4,eq5],[%c2,%c3,%c4]);
(%o12) [[%c2= $\frac{\rho R^2}{2 \text{eps}0}$ , %c3= $-\frac{\rho R^3}{3 \text{eps}0}$ , %c4=0]]
```

We define the integration constants and bring the potential in its final concise form:

```
(%i13) %c2:(rho*R^2)/(2*eps0)$ %c3:- (rho*R^3)/(3*eps0)$ %c4:0$
(%i16) phi_in(r);
(%o16)  $\frac{\rho R^2}{2 \text{eps}0} - \frac{r^2 \rho}{6 \text{eps}0}$ 
(%i17) phi_out(r);
(%o17)  $\frac{\rho R^3}{3 \text{eps}0 r}$ 
(%i18) Phi(r):=if r<=R then (rho*R^2)/(2*eps0)-(r^2*rho)/(6*eps0)
else (rho*R^3)/(3*eps0*r);
(%o18)  $\Phi(r):=if r \leq R \text{ then } \frac{\rho R^2}{2 \text{eps}0} - \frac{r^2 \rho}{6 \text{eps}0} \text{ else } \frac{\rho R^3}{3 \text{eps}0 r}$ 
```



Finally we calculate the field strength:

```
(%i21) E_in(r):=-diff(phi_in(r),r)$ E_out(r):=-diff(phi_out(r),r)$
(%i23) E_in(r);E_out(r);
(%o23)  $\frac{r \rho}{3 \text{eps}0}$ 
(%o24)  $\frac{\rho R^3}{3 \text{eps}0 r^2}$ 
(%i25) E(r):=if r<=R then (r*rho)/(3*eps0) else (rho*R^3)/(3*eps0*r^2);
(%o25)  $E(r):=if r \leq R \text{ then } \frac{r \rho}{3 \text{eps}0} \text{ else } \frac{\rho R^3}{3 \text{eps}0 r^2}$ 
```

Exercise: Substitute $\rho = \frac{Q}{\frac{4}{3}R^3\pi}$ and check if results of Method 2 meet the results of Method 1.

We plot for $Q = 1$, $\epsilon_0 = 1$ and $R = 2$.

```
(%i26) Q:1$ R:2$ eps0:1$
(%i29) rho:Q/(4/3*pi*R^3)$
(%i32) plot2d(Phi(r), [r, 0, 5]);
(%o32) [C:/Users/Josef/maxout.gnuplot]
(%i35) plot2d(E(r), [r, 0, 5]);
(%o35) [C:/Users/Josef/maxout.gnuplot]
```



Potential (above) and Strength (below)

Method Three

The solution can be found in a much easier way using *Gauss'* law. This is left to the reader.



II.20 A solid dielectrical sphere with an empty spherical cavity is charged uniformly with volume charge density ρ . The centre of the spherical cavity is placed at a distance r_0 from the centre of the solid. Find the strength of the electric field at any point inside the cavity.




Fig. II.20

Solution:

The spherical cavity can be treated as a uniformly charged sphere with zero volume charge density, which can be modelled as a superposition of a positive density ($+\rho$) and a negative density ($-\rho$) of equal magnitude. Then, by the superposition principle, the resultant electric field can be calculated as the vector sum of the fields produced by the whole solid sphere of charge density ($+\rho$) and the sphere of charge density ($-\rho$) corresponding to the cavity.

From the solution of Problem II.16, we know that inside a uniformly charged solid sphere the electric field is given by

$$\vec{E} = \frac{\rho}{3\epsilon_0} \vec{r}.$$

According to Fig. II.20, we can write

$$\vec{E} = \frac{\rho}{3\epsilon_0} \vec{R}_1 - \frac{\rho}{3\epsilon_0} \vec{R}_2 = \frac{\rho}{3\epsilon_0} (\vec{R}_1 - \vec{R}_2) = \frac{\rho}{3\epsilon_0} \vec{r}_0.$$



II.21 We consider a system of two conducting balls of radii r_1 and r_2 , separated by a large (in comparison with the radii) distance d . How should a charge Q be distributed between the balls, if the force acting between the balls is to decrease after the balls have been connected by a thin wire?

Solution:

Since the distance between the balls is much larger than the sum of their radii, the balls can be treated as point charges. Therefore, the force acting between them is of the form

```
(%i1) F: (Q-Q1)*Q1/(4*pi*d^2*eps0);
(%o1) (Q-Q1) Q1
      4 pi d^2 eps0
```

where Q_1 is the charge on the ball of radius r_1 and $(Q - Q_1)$ the charge on the other ball. After the balls have been connected by a thin wire, the charge is redistributed and the new force is

```
(%i2) F1: ((Q-q1)*q1)/(4*pi*d^2*eps0);
```

$$(\%o2) \frac{q_1(Q - q_1)}{4\pi d^2 \epsilon_0}$$

where q_1 and $(Q - q_1)$ are new charges.

When the balls are connected by a wire, then the potential difference disappears (the potentials on the balls are the same). Using the expression for the potential on a metallic ball of radius r charged with a charge q

```
(%i3) Phi(q, r) := 1/(4*pi*eps0)*(q/r);
```

$$(\%o3) \Phi(q, r) := \frac{1}{4\pi \epsilon_0} \frac{q}{r}$$

one can write the equation

$$\Phi(q_1, r_1) = \Phi(Q - q_1, r_2).$$

Charge q_1 is the solution of the above equation:

```
(%i4) solve(Phi(q1, r1)=Phi(Q-q1, r2), q1);
```

$$(\%o4) [q_1 = \frac{r_1 Q}{r_2 + r_1}]$$

```
(%i5) q1: (r1*Q)/(r2+r1)$
```

In order to find Q_1 we have to solve the following inequality

$$F_1 < F \text{ i.e. } F_1 - F < 0.$$

```
(%i6) factor(F1-F<0);
```

$$(\%o6) \frac{Q^2 - Q Q_1 + Q_1 Q - Q_1^2}{4\pi d^2 \epsilon_0} < 0$$

It can easily be seen that the above inequality is true within the zeros of a second order polynomial (parabola). Let's find these zeros now.

```
(%i7) ev(solve(F1-F=0, Q1));
```

$$(\%o7) [Q_1 = \frac{r_2 Q}{r_2 + r_1}, Q_1 = \frac{r_1 Q}{r_2 + r_1}]$$

From %o7 we can conclude that for $r_2 < r_1$ the solution is

$$\frac{Q r_2}{r_1 + r_2} < Q_1 < \frac{Q r_1}{r_1 + r_2}$$

and consequently when $r_2 > r_1$ then we have

$$\frac{Q r_1}{r_1 + r_2} < Q_1 < \frac{Q r_2}{r_1 + r_2}.$$

The screen shots on the next page demonstrate that DERIVE and TI-NspireCAS as well can solve the inequality $F_1 - F < 0$.

$[\epsilon_0 \in \text{Real}(0, \infty), d \in \text{Real}(0, \infty), r_1 \in \text{Real}(0, \infty), r_2 \in \text{Real}(0, \infty)]$

SOLVE(F1 - F < 0, Q1)

$$\frac{Q \cdot r_2}{r_1 + r_2} < Q_1 < \frac{Q \cdot r_1}{r_1 + r_2} \vee \frac{Q \cdot r_1}{r_1 + r_2} < Q_1 < \frac{Q \cdot r_2}{r_1 + r_2}$$

The solution of TI-NspireCAS is a "little bit" bulky and includes a lot of additional information. It should not be really funny to read this solution on a handheld.

$$\begin{aligned} & \text{solve}(f1-f<0,qq1)|d>0 \text{ and } \epsilon_0>0 \text{ and } r1>0 \text{ and } r2>0 \\ & \frac{qq \cdot r1}{r1+r2} < qq1 < \frac{qq \cdot r2}{r1+r2} \text{ and } d>0 \text{ and } r1>0 \text{ and } r2>0 \text{ and } \epsilon_0>0 \text{ or } qq1 > \frac{qq \cdot r1}{r1+r2} \text{ and } d>0 \text{ and } qq>0 \text{ and } r1+r2 > 0 \\ & qq1 > \frac{qq \cdot r1}{r1+r2} \text{ and } d>0 \text{ and } qq>0 \text{ and } r1+r2=0 \text{ and } r1>0 \text{ and } r2>0 \text{ and } \epsilon_0>0 \text{ or } \frac{qq \cdot r2}{r1+r2} < qq1 < \frac{qq \cdot r1}{r1+r2} \end{aligned}$$

II.22 The sketch shows two point charges Q and $2Q$. Ql charge is uniformly distributed on the segment of length l .




Fig. II.22

Find the angle α and relationship of Ql/Q charges for which the *Coulomb* force is disappearing in the centre of the coordinate system.

Solution:

Coulomb force is zero implies the electric field to be zero. Therefore it remains to inspect the electric field. The resultant field is a vector sum of the fields created by the point charges and the Ql charge. For the point charges we have

```
(%i1) assume(R>0,l>0)$ Q_:[Q,2*Q]$ r_: [0,0]$
(%i4) R_:R*matrix([cos(%pi/6),sin(%pi/6)],[-cos(alpha),sin(alpha)])$
(%i5) FQ_:1/(4*pi*eps0)*trigsimp(
      sum(Q_[j]/sqrt((r_-R_[j]).(r_-R_[j]))^3*(r_-R_[j]),
      j,1,length(Q_)));
(%o5) [(4*cos(alpha)-sqrt(3))/8*pi*eps0*R^2, (4*sin(alpha)+1)/8*pi*eps0*R^2]
```

and for the charge Ql we will write

```
(%i6) lambda:Ql/l$
(%i7) E1_: [0, -lambda/(4*pi*eps0)*integrate(1/y^2, y, R, R+1)];
(%o7) [0, -\frac{Ql \left( \frac{1}{R} - \frac{1}{R+1} \right)}{4 \pi \text{eps}0 l}]
```

Then the resultant field is the sum and will be evaluated applying the requested relationship of the charges $Ql = \eta \cdot Q$ (η is the factor of proportionality).

```
(%i8) Ftotal_:FQ_+E1_$
(%i9) Ftotal_:ev(Ftotal_,Ql=eta*Q);
(%o9) [\frac{(4 \cos(\alpha) - \sqrt{3}) Q}{8 \pi \text{eps}0 R^2}, -\frac{\eta Q \left( \frac{1}{R} - \frac{1}{R+1} \right) (4 \sin(\alpha) + 1) Q}{4 \pi \text{eps}0 l \cdot 8 \pi \text{eps}0 R^2}]
```

We try to find the desired parameters α and η in one single step.

(Remember, the `solve` function doesn't need the right hand side of the equation when it equals zero.)

```
(%i10) solve(Ftotal_, [alpha, eta]);
(%o10) []
```

As this does not work we will try to find the solution step by step (good old elimination method).

From the first equation we get

```
(%i11) solve(Ftotal_[1], alpha);
solve: using arc-trig functions to get a solution.
Some solutions will be lost.
(%o11) [\alpha = \arccos\left(\frac{\sqrt{3}}{4}\right)]
```

We know that trig equations in most cases have more solutions – let's try later with another CAS.

Now we will proceed finding η .

```
(%i12) solve(ev(Ftotal_[2], alpha=acos(sqrt(3)/4)), eta);
(%o12) [\eta = -\frac{(\sqrt{13}+1)R + (\sqrt{13}+1)l}{2R}]
(%i13) factor(%);
(%o13) [\eta = -\frac{(\sqrt{13}+1)(R+l)}{2R}]
```

Finally we are verifying the correctness of the obtained solution.

```
(%i14) factor(ev(Ftotal_, alpha=acos(sqrt(3)/4), eta=-((sqrt(13)+1)*(R+l))/(2*R)));
(%o14) [0, 0]
```

We will tackle this problem with TI-NspireCAS and DERIVE as well. TI-NspireCAS allows solving the system of equations in one step but the output is again more than bulky so that it might not be comfortable to read it off from the handheld. At the other hand TI-NspireCAS gives more than one solution.

$$\begin{aligned}
 & etotal_ := \frac{1}{4 \cdot \pi \cdot \epsilon 0} \cdot (row1 + row2) + e1_ | r > 0 \\
 & \left[\frac{qq \cdot (4 \cdot \cos(\alpha) - \sqrt{3})}{8 \cdot \epsilon 0 \cdot \pi \cdot r^2} \quad \frac{qq \cdot \eta}{4 \cdot l \cdot \epsilon 0 \cdot \pi \cdot (r+l)} - \frac{qq \cdot \eta}{4 \cdot l \cdot \epsilon 0 \cdot \pi \cdot r} - \frac{qq \cdot (4 \cdot \sin(\alpha) + 1)}{8 \cdot \epsilon 0 \cdot \pi \cdot r^2} \right] \\
 & solve(etotal_ [1,1]=0 \text{ and } etotal_ [1,2]=0, \{\alpha, \eta\}) \\
 & \leftarrow \text{or } r+l \neq 0 \text{ and } r \neq 0 \text{ and } \alpha = 2 \cdot n1 \cdot \pi - \cos^{-1}\left(\frac{\sqrt{3}}{4}\right) \text{ and } \epsilon 0 \neq 0 \text{ and } \eta = \frac{(\sqrt{13}-1) \cdot (r+l)}{2 \cdot r} \text{ or } r+l \neq 0 \text{ and } r \neq 0 \text{ and } \alpha = 2 \cdot n1 \cdot \pi
 \end{aligned}$$

So does DERIVE. Calculation procedure is not complicated and the output is very handsome, isn't it.

APPEND([[η, α]], SOLUTIONS(Etotal_, [η, α]))

$$\left[\begin{array}{cc}
 \eta & \alpha \\
 \frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} - \frac{1}{2}\right)}{R} & - \text{ACOS}\left(\frac{\sqrt{3}}{4}\right) \\
 \frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} - \frac{1}{2}\right)}{R} & \text{ASIN}\left(\frac{\sqrt{3}}{4}\right) + \frac{3 \cdot \pi}{2} \\
 - \frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} + \frac{1}{2}\right)}{R} & \text{ACOS}\left(\frac{\sqrt{3}}{4}\right)
 \end{array} \right]$$

$$\text{SUBST}\left(Etotal_, [\eta, \alpha], \left[\frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} - \frac{1}{2}\right)}{R}, - \text{ACOS}\left(\frac{\sqrt{3}}{4}\right) \right]\right) = [0, 0]$$

$$\text{SUBST}\left(Etotal_, [\eta, \alpha], \left[\frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} - \frac{1}{2}\right)}{R}, \text{ASIN}\left(\frac{\sqrt{3}}{4}\right) + \frac{3 \cdot \pi}{2} \right]\right) = [0, 0]$$

$$\text{SUBST}\left(Etotal_, [\eta, \alpha], \left[- \frac{(R+1) \cdot \left(\frac{\sqrt{13}}{2} + \frac{1}{2}\right)}{R}, \text{ACOS}\left(\frac{\sqrt{3}}{4}\right) \right]\right) = [0, 0]$$